

# 12. Building Blocks

## *General Description*

This remarkable program enables the user to create 3-D objects on a 2-D screen. It allows you to construct a picture with building blocks, whose positions are determined by the specification you give to the computer. Use the L key to expand your creation to the Left, R if you want it developed to the Right and so on. You can change the colour of the blocks by keying in the first letter of the colour you require, and if you want to see the computer draw the shape for you, simply press P for Playback.

## *Detailed Description*

**Line 10-20** First part of delete trick.

**30-100** Credits (can be deleted if 'no room').

**120** Set text window.

**130** MAX = Maximum number of operations, depends on size of memory, if 32K machine then MAX can be much higher.

**140-180** Set up variables.

**190** Clear graphics area.

**200-210** Jump to procedures to initialise cube and to display cube on the screen.

**220** Display for playback.

**230-360** Input from keyboard and display option in window.

**370-380** Jump to procedure to give help (i.e. list of commands).

**390-420** Add move to C\$(LT) so computer can play back picture later. Also sets colour of lines to colour chosen.

**430-610** Procedure to initialise cube.

**620-690** Procedure to display cube on screen.

**700-780** Procedure to display help in window.

**790-1070** Instructions.

**1080-1110** Second part of delete trick.

### *Educational Notes*

This program is valuable for its ability to bring geometric examples to life. Art teachers will find it useful as an aid to demonstrating the power and techniques of perspective, and computer students will waste hours trying to demonstrate that they have missed their vocation.

### *Program Listing*

```
10 ON ERROR GOTO 110
20 GOTO 790
30 REM *****
40 REM *          BUILDING BLOCKS          *
50 REM *
60 REM *          WRITTEN BY                *
70 REM *
80 REM *      Andrew Pusey. 1982            *
90 REM *
100 REM *****
110 MODE4
120 VDU28,0,30,15,22
130 MAX=150
140 DIMC$(MAX),XA(4),YA(4),XB(4),YB(4)
150 W=600:H=500
160 G=1:CL=3
170 M=300:Q=1:X=-0.5:B=4.5:Y=-0.5
180 LT=1:S=1
190 VDU16
200 PROC_INIT_SQUARE
210 PROC_PUT_SQUARE
220 IF LT<G THEN KB$=C$(LT):PRINT"}";:GOTO260
230 G=G+1
240 IF LT>MAX LT=MAX
250 PRINT"> ";:KB$=GET$
260 IFKB$="O"THENB=ABS(B-Q):PRINT"Out.":GOTO 390
270 IFKB$="I"THENB=B+Q:PRINT"In.":GOTO 390
280 IFKB$="L"THENY=Y-Q:PRINT"Left.":GOTO 390
290 IFKB$="R"THENY=Y+Q:PRINT"Right.":GOTO 390
300 IFKB$="U"THENX=X+Q:PRINT"Up.":GOTO 390
310 IFKB$="D"THENX=X-Q:PRINT"Down.":GOTO 390
320 IFKB$=" " THENXC=X+1:C$(LT)=KB$:LT=LT+1:PRINT"Space.":
GOTO220
330 IFKB$="P"THENG=G-1:PRINT"Playback.":GOTO180
340 IFKB$="Y"THENCL=3:PRINT"Colour yellow.":GOTO390
350 IFKB$="G"THENCL=2:PRINT"Colour green.":GOTO390
360 IFKB$="B"THENCL=4:PRINT"Colour blue.":GOTO390
370 PROC_HELP
380 GOTO250
390 C$(LT)=KB$:LT=LT+1
```

```

400 XC=XC-1:IFXC>.5THEN220
410 VDU19,1,CL,0,0,0
420 GOTO200
430 DEF PROC_INIT_SQUARE

440 XA(0)=Y/B
450 XA(1)=XA(0)
460 XA(2)=(Y+Q)/B
470 XA(3)=XA(2)
480 XA(4)=XA(0)
490 YA(0)=X/B
500 YA(1)=(X+Q)/B
510 YA(2)=YA(1)
520 YA(3)=YA(0)
530 YA(4)=YA(0)
540 BB=B/(B+Q):XC=1
550 FORN=0TO4
560   XA(N)=XA(N)*M
570   XB(N)=XA(N)*BB
580   YA(N)=YA(N)*M
590   YB(N)=YA(N)*BB
600 NEXT
610 ENDPROC
620 DEF PROC_PUT_SQUARE
630 FOR N=0TO3
640   PLOT69,XA(N+1)+W,YA(N+1)+H
650   DRAWXA(N)+W,YA(N)+H
660   DRAWXB(N)+W,YB(N)+H
670   DRAWXB(N+1)+W,YB(N+1)+H
680 NEXT
690 ENDPROC
700 DEF PROCHELP
710 CLS
720 PRINT"R..Right "
730 PRINT"L..Left  "
740 PRINT"U..Up    "
750 PRINT"D..Down  "
760 PRINT"I..In    "
770 PRINT"O..Out   "
780 ENDPROC
790 MODE7
800 PRINTTAB(10,4);CHR$(141);CHR$(130);"BUILDING BLOCKS."
810 PRINTTAB(10,5);CHR$(141);CHR$(130);"BUILDING BLOCKS."
820 PRINTTAB(7,8);CHR$(129);"INSTRUCTIONS."
830 PRINT'" This program is designed to allow"
840 PRINT"you to make 3-D objects on a 2-D screen."
850 PRINT"You use blocks to build the object"
860 PRINT"which you can move up, down, left,"
870 PRINT"right, in and out."
880 PRINT"You can change the colour of your"
890 PRINT"object by pressing 'G' for green, 'B'"
900 PRINT"for blue and 'Y' for yellow."
910 PRINT CHR$(136);CHR$(129);"      Press any key to contin
ue."
920 IF GET$=""THEN920
930 CLS
940 PRINT'" To move, you press the first letter of"
950 PRINT"that direction, e.g."
960 PRINT'"To go left press 'L', or"
970 PRINT"To go in press 'I'."
980 PRINT'"If you wish to see the computer draw"
990 PRINT"your object then press 'P' for playback"
1000 PRINT"and the computer will draw it at the"
1010 PRINT"point where you stopped drawing."
1020 PRINT"There is a limit to how much the"
1030 PRINT"computer will remember and this is about";
1040 PRINT"150 moves."
1050 PRINT'"Are you sure you wish to continue with this p
rogram (Y/N) ?";

```

```
1060 IFGET$<>"Y"THEN END
1070 CLS:PRINTTAB(2,10);
1080 PRINT"PRESS FUNCTION KEY 'f0'"
1090 *KEY0 DELETE790,1200|MRUN|M
1100 IF INKEY(-33)=0THEN 1100
1110 END
```