
26. Printer Wordgame

General Description

This is the user defined version of wordsquare. You may select
the words that you wish to have hidden in the wordsquare and
then the computer will generate the wordsquare and dump it to
your printer. You may then photocopy it and print it. I run a
request service for this at school and it satisfies the requests from
other departments admirably. Of course this does presuppose
that you have a printer. If not then you will have to save this
program till you acquire one. You can always use the next
program meanwhile. Note that the printer defaults to a serial
Epson set at 9600 baud with line feed not activated. These
printer settings are at lines 70, 80 and 90. If you have a parallel
printer then change line 70 to *FX5,1 - you will have to decide if
you wish to suppress any characters as in line 90. The BBC User
guide helps here.

Details Description

Lines 10-172 Clear variables. Lines 70 to 90 sets up printer.
(These values may be changed depending on what sort of printer
is being used.) In this section of the program, the cursor is turned
off and the cursor keys disabled. Procedure calls.

180-240 Save array spaces and give X(N), Y(N) directional
operations - e.g. vertical, horizontal, backwards etc. (eight
different directions).

250-570 User to enter up to twenty words, each no more than
seventeen characters in length. Errors can be deleted in the usual
way. Only uppercase alpha characters will be accepted.

580-940 Prints user data to screen.
950-1050 Prints grid to screen.

Educational Notes

See comments on next wordsquare program.

Program Listing

   10  REM *****************************
   20  REM ** WORDS                   **
   30  REM ** Written by Ian Clarke.  **
   40  REM ** Feb 83                  **
   50  REM *****************************
   60  CLEAR
   70  *FX5,2
   80  *FX8,7
   90  *FX6,0
  110  MODE 7
  111  VDU 23;8202;0;0;0
  112  PRINT
  113  *FX4,1
  114  *FX11,0
  120  PROCarr
  130  PROCinput
  140  PROCsqr
  150  PROCdata
  160  PROCprint
  167  *FX4 
  168  *FX11,25
  169  *FX15
  170  A=GET
  171  MODE 7
  172  END
  173  REM -----------------------------
  174  REM     Set up arrays.
  180  DEFPROCarr
  190  DIM C$(20,20),X(8),Y(8),W$(21)
  200  X(1)=1:X(2)=1:X(3)=0:X(4)=-1
  210  X(5)=-1:X(6)=-1:X(7)=0:X(8)=1
  220  Y(1)=0:Y(2)=1:Y(3)=1:Y(4)=1
  230  Y(5)=0:Y(6)=-1:Y(7)=-1:Y(8)=-1
  239  REM .............................
  240  ENDPROC
  241  REM -----------------------------
  242  REM  Input the words the user
  243  REM  wants printed on a grid.
  250  DEFPROCinput
  260  CLS
  270  FOR C=1 TO 20
  280    X=0
  290    PRINT TAB(10,15);CHR$(141);CHR$(129);"Enter word no
 ";C
  300    PRINT TAB(10);CHR$(141);CHR$(129);"Enter word no ";
C
  310    REPEAT
  320      X=X+1
  330      *FX15
  340      A$=GET$
  350      IF ASC(A$)<>127 THEN GOTO 390
  360      IF LEN(W$(C))>1 THEN X=X-2 ELSE X=0
  370      W$(C)=LEFT$(W$(C),LEN(W$(C))-1)

  380      GOTO 420
  390      IF ASC(A$)=13 THEN GOTO 480
  400      IF A$<"A" OR A$>"Z" THEN GOTO 330
  410      W$(C)=W$(C)+A$
  420      PRINT TAB(8,17);CHR$(141)SPC40
  430      PRINT TAB(8);CHR$(141)SPC40
  440      PRINT TAB(8,17);CHR$(141);CHR$(129);W$(C)
  450      PRINT TAB(8);CHR$(141);CHR$(129);W$(C)
  460    UNTIL X=17
  470    GOTO 490
  480    IF W$(C)="" THEN C1=C-1:GOTO 570
  490    CLS
  500  NEXT C
  510  C1=20
  560  REM .............................
  570  ENDPROC
  571  REM -----------------------------
  572  REM    Thinks of grid for
  573  REM    chosen words.
  580  DEFPROCsqr
  590  FOR N%=1 TO C1
  600    W$=W$(N%)
  610    Z%=RND(8)
  620    Y%=Y(Z%) : X%=X(Z%)
  630    X4%=RND(20):Y4%=RND(20)
  640    U$=C$(X4%,Y4%)
  650    IF U$="" THEN GOTO 730
  660    I%=0
  670    REPEAT
  680      I%=I%+1
  690    IF MID$(W$,I%,1)=U$ THEN X4%=X4%-((I%-1)*X%):Y4%=Y4
%-((I%-1)*Y%):UNTIL I%>0:GOTO 720
  700  UNTIL I%=LEN(W$)
  710  GOTO 610
  720  IF X4%<1  OR X4%>20 OR Y4%<1 OR Y4%>20 THEN GOTO 610
  730  X1%=X4%:Y1%=Y4%
  740  W%=0
  750  REPEAT
  760    W%=W%+1
  770  IF C$(X1%,Y1%)<>"" AND C$(X1%,Y1%)<>MID$(W$,W%,1) THE
N UNTIL W%>0:GOTO 610
  780  X1%=X1%+X%:Y1%=Y1%+Y%

  790  IF X1%<1 OR X1%>20 OR Y1%<1 OR Y1%>20 THEN UNTIL W%>0
:GOTO 610
  800  UNTIL W%=LEN(W$)
  810  X1%=X4%:Y1%=Y4%
  820  FOR K%=1 TO LEN(W$)
  830    C$(X1%,Y1%)=MID$(W$,K%,1)
  840    X1%=X1%+X%:Y1%=Y1%+Y%
  850  NEXT K%
  860  NEXT N%
  869  REM .............................
  870  ENDPROC
  871  REM -----------------------------
  872  REM    Print the user's words
  873  REM    to the printer.
  880  DEFPROCdata
  890  VDU 2
  900  CLS
  910  FOR Q%=1 TO C1 STEP 2
  920    PRINT W$(Q%);TAB(20);W$(Q%+1)
  930  NEXT Q%
  931  FOR P=1 TO 5000:NEXT
  932  REM .............................
  940  ENDPROC
  941  REM -----------------------------
  942  REM     Print grid to printer.
  950  DEFPROCprint

  960  FOR Y%=1 TO 20
  970    FOR X%=1 TO 20
  980      IF C$(X%,Y%)<>"" THEN GOTO 1000
  990      C$(X%,Y%)=CHR$(RND(26)+64)
 1000      PRINT TAB((X%-1)*2,Y%);CHR$(133);C$(X%,Y%);SPC40
 1010    NEXT X%
 1020    PRINT
 1030  NEXT Y%
 1040  VDU 3
 1041  REM ............................
 1050  ENDPROC

