
15. Draw and Save

General Description

This program was written for use in the classroom and
laboratory, and is designed for the teaching and practice of
electronics. Initially, it simply enables circuit diagrams to be
drawn on the TV screen, more effectively than on a blackboard
and just as fast. More usefully, however, once drawn, it allows
the diagrams to be recorded on tape ready to be instantly
recalled. Thus a library of teaching circuits is always readily
available, and individual projects can be easily preserved.

The program allows a 'blip' to be moved about the screen
using the cursor keys, and this 'lays' the symbols as required - the
procedures PROCRES, PROCAP etc. each drawing a different
symbol and each being called by a different key (C, R, etc). The
coordinates of the endpoints of the last symbol are chosen either
(i) to be the endpoint of the last symbol and where the blip has
been moved to before C, R or whatever has been pressed, or (ii)
by pressing key B when the blip has been moved to one chosen
point and C, R, etc when it is at the other.

The procedures only permit horizontally or vertically aligned
endpoints, with the exception of PROCLINE which connects in
any direction. Lettering and numbering is done by first pressing
key @. This calls up the appropriate procedure, so that keys
simply print their face values until @ is pressed again.

Each time a symbol or letter is 'laid', its type and coordinates
are stored in an array, and when a circuit is completed it may be
given a title and recorded on tape (or recalled) by the appropriate
procedure. (BPUT# was used rather than PRINT# because the
operating system had a bug which was very troublesome when
handling this type of data.)

The top six or seven lines on the screen are used for
instructions (when requested), recording messages and so forth,
and although they may be easily cleared, to do so also clears any

drawing in that region (but not, of course, from memory). In fact
there are sufficient descriptions and instructions included as part
of the program to enable it to be used by someone who has not
read these notes, which accounts for its considerable length.

One capability not mentioned in the program is that the blip
may be made just to draw lines on the screen as it is moved
around. This is switched on by key D and off by B. These lines
are not recorded. No provision has been included to enable the
user to easily erase a wrongly drawn symbol, although this can
somewhat laboriously be achieved by printing spaces all over it!

The layout of the drawing procedures are similar enough to
enable more symbols to be added to the program's capability.

Note that the hash symbol (#) prints as £ with this printer.
'BPUT£' means 'BPUT#', etcetera.

Detailed Description

Line 40 Sets up variables and arrays used.
LAT (300, 1) Coordinates of letters. LAB$ (300) Letters.
COORD (50, 4) Type of symbol and its coordinates.

50-60 Sets colours used.
70-150 Description and instructions.
180-220 Set up text window, flush key buffer, disable cursor

editing keys (so they can be used to move the blip).
230-350 Uses GET and IF-THEN statement to test keyboard

and call appropriate procedures.
360 Erase blip (at coordinate P, Q).
370-400 Scan cursor keys to move blip (in steps of 30).
420 Replace blip at same or new position (280 and 410 draw

lines on the screen as blip moves).
430 Return to 230 and scan keys again.

PROCRES - This procedure will be examined in detail

320 If R is pressed (ASCII-82), then K=P: L=Q PROCRES
(H, J, K, L) P & Q are the coordinates of the blip (drawn by 420)
H & J - the coordinates of the other end were set earlier either by
line 290 (if B was pressed earlier), or at the end of the last
drawing procedure (see later).

460 PROCRES(A, B, C, D) - Simply substitutes A=H, B=J,

C=K, D=L.
470 Local variables.
490 COORD is a 50 x 5 array. It may be thought of as 50

rows corresponding to 50 symbols. Each row has 5 elements -
the first identifies the symbol - 1 for a resistor, 2 for a capacitor
and so on. The other 4 are the X, Y coordinates of the ends of
that symbol. This line is therefore recording the coordinates in
memory.

500 N = N + 1 so the next symbol will occupy the next row.
510 Necessary trap (in case a key is pressed for too long).

This line would be better moved to 485 so as not to waste array
space.

530 In the actual drawing of a horizontal register it will be
assumed that C>A. If it were not, this would result in an odd
symbol. This line swops C and A if necessary, and 'remembers'
that it has done so by setting flag F=1.

540 As for 530 for vertical resistors.
550 Vertical resistors.
560 Horizontal resistors.
570-580 Trap for others.
600-680 Draws vertical symbols.
690-710 Draws horizontal symbols.
730 Sets the coordinates of the next symbol to the end of this

one.
740 Moves the blip to the correct end if misplaced by line

530.
760-810 PROCLINE - as PROCRES but simpler (but lines).
820-1060 PROCCAP - as PROCRES (but capacitors).
1070-1240 PROCTRN - as PROCRES but (i) only draws

vertical transistors and (ii) in order that the base should link
neatly with the other components it must be centred on a
possible line (the blip moves in steps of 30 - line 400).

1250-1460 PROCDIO - Draws DIODES (called by I not D).
1470 PROCLET - Lettering.
1490 Print text at graphics cursor.
1500-1570 REPEAT-UNTIL loop - Flushes keyboard buffer,

waits until key is pressed, stores coordinates of the blip in the
300 × 2 array LAT and the actual key pressed in the 300 x 1
array LAB$ - Increments the X coordinate (P) by 30 and the
array rows (LL) by 1. Line 1570 checks if the key pressed was
the terminator.
82



1580-1600 Sends text cursor back to text window and flushes
the keyboard buffer.

1610 PROCSAVE - Saves diagrams to tape.
1640-1660 Opens channel to tape.
1670 Sends the number of symbols.
1680-1770 FOR-NEXT loop - Sends type and coordinates of

each symbol. (These are in fact divided by 10 because BPUT#
requires numbers below 255 (FF in Hex).

1780-1880 REPEAT-UNTIL loop - Sends coordinates of each
letter (again divided by 10) and its ASCII code, then closes the
channel.

1900 PROCBACK - Fetches diagrams from tape.
1910-1950 Options title of required diagram - opens that

channel.
1960-2040 FOR-NEXT loop - accepts first of symbols

(compare line 1670) then the type and coordinate of each which
are stored in the same 50 x 5 array COORD (after being
multiplied by 10).

2050-2150 REPEAT-UNTIL loop - accepts coordinates of
lettering and stores in LAT and LAB$.

2160-2240 REPEAT-UNTIL loop - prints symbols.
2250-2320 REPEAT-UNTIL loop - prints lettering.
2340-2370 PROCMENU - Displays instructions.

Educational Notes

This program is largely a demonstration program for the teacher.
Children have experimented with it but find that is command
structure is difficult to use. It needs a large colour monitor.

It can be used to generate and save a variety of simple line
drawings and not just circuit diagrams. It loads the data quickly
from tape and thus if you, as a teacher, can think far enough
ahead to generate a data tape with all your different diagrams in
sequence on the tape, as a classroom tool it will prove very
useful.

Program Listing

   10 MODE4

   20 REM R.V.Woodworth,Holloway School
   30 REM *********************************GLOBAL VARIABLES-
COLOURS-INSTRUCTIONS*
   40 DIM LAT(300,1):DIM LAB$(300):DIM COORD(50,4):N=0:Z=0:L
=0:F=0:LL=0:U=1:P=0:Q=0:R=0:A=0:B=0:C=0:M=1:W=0:D=0:H=0:J=0:
K=0:L=0:U=1:Z=1
   50 VDU19,0,4,0,0,0
   60 VDU19,7,3,0,0,0
   70 PRINT TAB(12,7)"DRAW & SAVE"
   80 PRINT TAB(15,8)"BY"
   90 PRINT TAB(4,10)"R.V.Woodworth, Holloway School"''"  Th
is program draws circuit diagrams."'"  The blip is moved usi
ng the cursor "
  100 PRINT"keys and 'lays' symbols to order:- R for";"resis
tor, C for capacitor etc."''"  Each will be connected to the
"'"previous one unless a different point"
  110 PRINT"is 'laid' by pressing B. (And B must be"'"presse
d at the start of the first"'"symbol and after lettering.)"'
  120 PRINT" Note. Symbols (except 'lines') must be"'"vertic
al or horizontal."'
  130 PRINT"  The lettering 'mode' is entered and "'"left by
 pressing @."''"  Circuits may be saved to tape by"'"pressin
g J and any recalled by K."'
  140 PRINT"Instructions may be called at any "'"time by pre
ssing H (for Help) BUT they"'"will overwrite the top 6 lines
 which"'"should not therefore be drawn on."
  150 PRINT'"Now press C (&RETURN) to continue"
  160 INPUTC$
  170 IF C$="C" CLS ELSE 160
  180 REM********************************TEXT WINDOW-FLUSH B
UFFER-DISABLE CURSOR
  190 VDU 28,0,6,39,0
  200 *FX 15,1
  210 *FX4,1
  220 REM*********************************** SCAN KEYS-SELEC
T OPERATION *******
  230 WW=GET
  240 IF WW=87 THEN CLS
  250 IF WW=74 THEN PROCSAVE
  260 IF WW=75 PROCBACK
  270 IF WW=72 THEN PROCMENU
  280 IF WW=68 THEN M=0
  290 IF WW=66 THEN M=1:H=P:J=Q
  300 IF WW=64 THEN PROCLET
  310 IF WW=76 THEN K=P:L=Q:PROCLINE(H,J,K,L)
  320 IF WW=82THEN K=P:L=Q:PROCRES(H,J,K,L)
  330 IF WW=67 THEN K=P:L=Q:PROCCAP(H,J,K,L)
  340 IF WW=84THENK=P:L=Q:PROCTRN(H,J,K,L)
  350 IF WW=73 THENK=P:L=Q:PROCDIO(H,J,K,L)
  360 PLOT67,0,0
  370 IF WW=136 THEN P=P-30
  380 IF WW=137 THEN P=P+30
  390 IF WW=138 THEN Q=Q-30
  400 IF WW=139 THEN Q=Q+30
  410 IF M=0 THEN DRAWP,Q
  420 MOVEP,Q:PLOT65,0,0
  430 GOTO 230
  440 END
  450 REM****************************************** PROCEDUR
ES ****************
  460 DEF PROCRES(A,B,C,D)
  470 LOCAL T,F,G,S
  480 REM*********************************NEW OR SAVE?-VERT.
OR HORIZ.etc.****
  490 COORD(N,0)=1:COORD(N,1)=A:COORD(N,2)=B:COORD(N,3)=C:CO
ORD(N,4)=D
  500 N=N+1
  510 IF A=C AND B=D THEN GOTO 750
  530 IF A>C THEN S=A:A=C:C=S:F=1

  540 IF B>D THEN T=B:B=D:D=T:G=1
  550 IF A=C GOTO 600
  560 IF B=D GOTO 690
  570 PRINT"ERROR - THIS PROGRAM ONLY ACCEPTS "'"VERTICAL OR
 HORIZONTAL RESISTORS"'"REDEFINE SECOND POINT (R) OR BOTH"
  580 GOTO 750
  590 REM***DRAW SYMBOL-(VERT or HORIZ)**
  600 MOVEA,B
  610 DRAWA,(B+D-100)/2
  620 DRAW(A+10),(B+D-100)/2
  630 DRAW(A+10),(B+D+100)/2
  640 DRAW(A-10),(B+D+100)/2
  650 DRAW(A-10),(B+D-100)/2
  660 MOVEA,(B+D+100)/2
  670 DRAWA,D
  680 GOTO 730
  690 MOVEA,B
  700 DRAW(A+C-100)/2,B:DRAW(A+C-100)/2,B+10:DRAW(A+C-100)/2
,B-10:DRAW(A+C+100)/2,B-10
  710 DRAW(A+C+100)/2,B+10:DRAW(A+C-100)/2,B+10:MOVE (A+C+10
0)/2,B:DRAWC,D
  730 H=K:J=L
  740 IF F=1 MOVE A,D:IF G=1 MOVE C,B
  750 ENDPROC
  760 REM********************************************** LINE
 *******************
  770 DEF PROCLINE(A,B,C,D)
  780 COORD(N,0)=5:COORD(N,1)=A:COORD(N,2)=B:COORD(N,3)=C:CO
ORD(N,4)=D:N=N+1
  790 MOVEA,B:DRAWC,D
  800 H=K:J=L
  810 ENDPROC
  820 REM******************************************* CAPACIT
ORS *****************
  830 DEF PROCCAP(A,B,C,D)
  840 LOCAL F,G,S,T
  850 REM***********************************NEW or SAVED-VER
T or HORIZ etc.******
  860 COORD(N,0)=2:COORD(N,1)=A:COORD(N,2)=B:COORD(N,3)=C:CO
ORD(N,4)=D:N=N+1
  870 IF A=C AND B=D GOTO 1060   
  880 IF A>C THEN S=A:A=C:C=S:F=1
  890 IF B>D THEN T=B:B=D:D=T:G=1
  900 IF A=C GOTO 950
  910 IF B=D GOTO 1000
  920 PRINT"ERROR - THIS PROGRAM ONLY ACCPETS "'"VERTICAL OR
 HORIZONTAL CAPACITORS"'"REDEFINE SECOND POINT (R) OR BOTH"
  930 GOTO 1060
  940 REM********* DRAW SYMBOL **********
  950 MOVEA,B:DRAWA,(B+D-20)/2
  960 DRAWA-40,(B+D-20)/2:DRAWA+40,(B+D-20)/2
  970 MOVEA+40,(B+D+20)/2:DRAWA-40,(B+D+20)/2
  980 MOVE A,(B+D+20)/2:DRAWA,D
  990 GOTO 1030
 1000 MOVE A,B
 1010 DRAW(A+C-20)/2,B:DRAW(A+C-20)/2,B-40:DRAW(A+C-20)/2,B+
40:MOVE(A+C+20)/2,B+40
 1020 DRAW(A+C+20)/2,B-40:MOVE(A+C+20)/2,B:DRAWC,D
 1030 H=K:J=L
 1040 IF F=1 MOVE A,D
 1050 IF G=1 MOVE C,B
 1060 ENDPROC
 1070 REM***************************************** TRANSISTO
RS ****************
 1080 DEF PROCTRN(A,B,C,D)
 1090 LOCALU,G,P
 1100 REM*********************************** NEW or SAVED et
c.*****************
 1110 COORD(N,0)=3:COORD(N,1)=A:COORD(N,2)=B:COORD(N,3)=C:CO

ORD(N,4)=D:N=N+1
 1120 IF A=C AND B=D GOTO 1240
 1130 IF B>D THEN P=B:B=D:D=P:G=1
 1140 IF A=C GOTO1180
 1150 PRINT"ERROR - THIS PROGRAM ONLY ACCEPTS"'"VERTICAL TRA
NSISTORS."
 1160 GOTO 1240
 1170 REM******* DRAW SYMBOL ************
 1180 U=B+((30*(INT((D-B)/60))))
 1190 MOVEA,B:DRAWA,(U-70):DRAWA-10,(U-40):MOVEA-20,U-70:DRA
WA,(U-70)
 1200 DRAWA-80,U:DRAWA-80,U+30:DRAWA-80,U-30:DRAWA-80,U
 1210 DRAWA,U+70:DRAWA,D
 1220 H=K:J=L
 1230 IF G=1 THEN MOVE C,B
 1240 ENDPROC
 1250 REM********************************************* DIODE
S ***************** 
 1260 DEF PROCDIO(A,B,C,D)
 1270 LOCAL U,V,G
 1280 REM************************************ NEW or SAVED-V
ERT or HORIZ etc.****
 1290 COORD(N,0)=4:COORD(N,1)=A:COORD(N,2)=B:COORD(N,3)=C:CO
ORD(N,4)=D:N=N+1
 1300 IF A=C AND B=D GOTO 1460
 1310 IF B=D GOTO1360
 1320 IF A=C GOTO 1400
 1330 PRINT"ERROR: THIS PROGRAM ONLY ACCEPTS"'"VERTICALORHOR
IZONTAL COMPONENTS"'"REDEFINE SECOND POINT(D) OR BOTH"
 1340 GOTO 1460
 1350 REM************* DRAW SYMBOL *******
 1360 U=(A+C)/2
 1370 IFC>A V=(A+C+50)/2 ELSE V=(A+C-50)/2
 1380 MOVE A,B:DRAWU,B:DRAWU,B+25:DRAWU,B-25:DRAWV,B:MOVEU,B
+25:DRAWV,B:DRAWV,B+25:DRAWV,B-25:DRAWV,B:DRAWC,D
 1390 GOTO 1450
 1400 MOVEA,B
 1410 U=(B+D)/2
 1420 IF D>B V=(B+D+50)/2 ELSE V=(B+D-50)/2
 1430 DRAWA,U:DRAWA-25,U:DRAWA,V:DRAWA+25,U:DRAWA,U
 1440 MOVE A-25,V:DRAWA+25,V:MOVEA,V:DRAWC,D
 1450 H=K:J=L
 1460 ENDPROC
 1470 REM******************************************* LETTERI
NG **************** MOVE CURSOR-NEW or SAVED-PRINT-etc.***
 1480 DEF PROCLET
 1490 VDU5
 1500 REPEAT
 1510   *FX 15,1
 1520   P$=GET$
 1530   LAT(LL,0)=P:LAT(LL,1)=Q:LAB$(LL)=P$
 1540   IF P$="@" GOTO 1550 ELSE PRINTP$
 1550   P=P+30:MOVEP,Q
 1560   LL=LL+1
 1570 UNTIL P$="@"
 1580 VDU4
 1590 *FX 15,1
 1600 ENDPROC
 1610 REM**************************************** SAVE TO TA
PE ******************
 1620 DEF PROCSAVE
 1630 LOCAL R,Y,U,I,O,G,L,B,C,D$,M
 1640 INPUT"Please type the name under which this    circuit
 is to be filed (then RETURN)",F$
 1650 REM************************************* OPEN FILE-SEN
D COORDINATES etc.***
 1660 X=OPENOUT F$
 1670 XX=N:BPUT#X,XX
 1680 PRINT"RECORDING"

 1690 FOR M=0TO N
 1700   Y=(COORD(M,0)):U=(COORD(M,1))/10:I=(COORD(M,2))/10:O
=(COORD(M,3))/10:G=(COORD(M,4))/10
 1710   IFU<0 U=0
 1720   IF I<0 I=0
 1730   IFO<0 O=0
 1740   IFG<0 G=0
 1750   BPUT#X,Y:BPUT#X,U:BPUT#X,I:BPUT#X,O:BPUT#X,G
 1760   VDU13:PRINTM;
 1770 NEXTM
 1780 REM***************************************** SEND LETT
ERING *************** 
 1790 REPEAT
 1800   B=(LAT(L,0))/10:C=(LAT(L,1))/10
 1810   D$=LAB$(L):R=ASC(D$)
 1820   BPUT#X,B:BPUT#X,C
 1830   BPUT#X,R
 1840   VDU13:PRINTL;
 1850 L=L+1:UNTIL L=LL+1
 1860 CLOSE#X
 1870 PRINT"Circuit ";F$;" has now been recorded." :PRINT"ST
OP TAPE"
 1880 ENDPROC
 1890 REM***************************************************
*************************** CALL FROM TAPE******************
*
 1900 DEF PROCBACK
 1920 INPUT"Please type the name of the circuit       requir
e (then press RETURN)",G$
 1930 PRINT"RUN TAPE"
 1940 REM************************************** OPEN CHANNEL
-FETCH COORDINATES***
 1950 T=OPENING$
 1960 XX=BGET#T:PRINTXX
 1970 FOR W=0TO XX
 1980   R=BGET#T:COORD(W,0)=R
 1990   F=BGET#T:COORD(W,1)=F*10
 2000   V=BGET#T:COORD(W,2)=V*10
 2010   G=BGET#T:COORD(W,3)=G*10
 2020   Y=BGET#T:COORD(W,4)=Y*10
 2030   VDU13:PRINTW;
 2040 NEXTW
 2050 REM***************************************** FETCH LET
TERING **************
 2060 REPEAT
 2070   B=BGET#T:LAT(LL,0)=B*10
 2080   C=BGET#T:LAT(LL,1)=C*10
 2090   S=BGET#T:D$=CHR$(S):LAB$(LL)=D$
 2100   VDU13:PRINTLL;
 2110   LL=LL+1
 2120 UNTIL EOF#T
 2130 CLOSE#T
 2140 PRINT" COMPLETED - STOP TAPE"
 2150 REM*********************************** CLOSE CHANNEL-P
LOT DATA ************ 
 2160 REPEAT
 2170   F=COORD(M,1):V=COORD(M,2):G=COORD(M,3):Y=COORD(M,4)
 2180   IF COORD(M,0)=1 PROCRES(F,V,G,Y)
 2190   IF COORD(M,0)=2 PROCCAP(F,V,G,Y)
 2200   IF COORD(M,0)=3 PROCTRN(F,V,G,Y)
 2210   IF COORD(M,0)=4 PROCDIO(F,V,G,Y)
 2220   IF COORD(M,0)=5 PROCLINE(F,V,G,Y)
 2230 M=M+1:UNTIL M=XX+1
 2240 REM****************************************** PLOT LET
TERING **************
 2250 REPEAT
 2260   B=LAT(K,0):C=LAT(K,1):D$=LAB$(K)
 2270   IF D$="@" GOTO 2290
 2280   VDU5:MOVEB,C:PRINTD$

 2290   K=K+1
 2300 UNTIL K=LL
 2310 VDU4
 2320 ENDPROC
 2330 REM*********************************** INSTRUCTIONS **
*********************
 2340 DEF PROCMENU
 2350 PRINT"Cursor keys (arrows!) move 'blip';"'"B=BLIP (To 
start component); L=Line"'"R=Resistor; C=Capacitor; T=Transi
stor; "'"I=Diode; @=Start or stop lettering;"'"H=Instruction
s; W=Wipe them off"
 2360 PRINT"J=SAVE this drawing; K=Fetch another"
 2370 ENDPROC

