
34. Maths Stars

General Description

This program produces various mathematical stars within a
circle, starting with three points on the circle and increasing the
number. It is possible to stop the demonstration by pressing any
key, and to stop the program using the ESC key. The program
runs in Mode 5 on a Model A, but could be enhanced for the
Model B.

Detailed Description

Lines 10-300 This is the main structure of the program. See
the user manual (page 77) for the disabling of the cursor in line
160. Colours are activated here.

310-390 This procedure saves the points on the edge of the
circle for the current number of points (N).

410-520 Draws the lines to the plotted points, calling a
routine to allow you to freeze the display.

530-end This stops the display for a short while, calling the
start/stop procedure.

Educational Note

Drawing these 'roses' is a favourite end of term pastime, and as
such it is now possible to add a little spice to this. Many children
can cope with drawing the simple roses but find the ones with
several points very difficult to master. Usually their patience
runs out before the pencil lead does. This will at least show them
quickly what it was intended that they achieve.

Program Listing

>;LIST
   10  REM ******************************
   20  REM **          STARS           **
   30  REM **    PROGRAM WRITTEN BY    **
   40  REM **      BY Ian Clarke       **
   50  REM **         Jan 83           **
   60  REM ******************************
   70  ON ERROR GOTO 710
   80  MODE 5
   90  CLS
  100  DIM X(20),Y(20)
  110  X1=640:Y1=512
  120  ENVELOPE 1,1,2,-2,0,10,20,1,1,0,0,-2,90,90
  130  N=3
  140  GCOL 0,3
  150  VDU 19,0,2,0,0,0
  160  VDU 23;8202;0;0;0;
  170  PRINT
  180  VDU 19,1,7,0,0,0
  190  PRINT TAB(8,4);"STARS."
  200  PRINT TAB(1);"You can start/"
  210  PRINT TAB(1);"stop this program"
  220  PRINT TAB(1);"by pressing any"
  230  PRINT TAB(1);"of the keys."
  240  COLOUR 2
  250  Q=INKEY(800)
  260  PROCposition
  270  PROCdrawlines
  280  N=N+1
  290  IF N=21 THEN GOTO710
  300  GOTO 260
  310  REM ------------------------------
  320  DEF PROCposition
  330  REM SAVES THE POSITIONS OF THE CIRCLE.
  340  CLG
  350  FOR I=1 TO N
  360    X(I)=480*COS(ATN(1)*8*I/N)
  370    Y(I)=506*SIN(ATN(1)*8*I/N)
  380  NEXT
  390  ENDPROC
  400  REM ------------------------------
  410  DEF PROCdrawlines
  420  REM DRAWS LINES TO EACH POINT OF THE CIRCLE
  430  PRINT TAB(2,28);N
  440  PRINT TAB(0,30);"POINTS"
  450  FOR I=1 TO N-1
  460    FOR C=1 TO N-I
  470      MOVE X(C)+X1,Y(C)+Y1
  480      DRAW X(C+I)+X1,Y(C+I)+Y1
  490      PROCstartstop
  500    NEXT C
  510  NEXT I
  520  REM ..............................
  530  REM START/STOP THE PROGRAM
  540  Q=0
  550  REPEAT
  560    Q=Q+1
  570    PROCstartstop
  580  UNTIL Q=150
  590  ENDPROC
  600  REM ------------------------------
  610  DEFPROCstartstop
  620  F$=INKEY$(1)
  630  IF F$=""THEN ENDPROC

  640  SOUND 1,1,120,255
  650  F=GET
  660  Q=149
  670  SOUND &1011,0,0,0
  680  FORX=1TO1000:NEXT
  690  ENDPROC
  700  REM ..............................
  710  CLS
  720  PRINTTAB(8,15);"BYE !!!"
  730  FORX=1TO5000:NEXT
  740  CLS
  750  END

