

25. Meter Reader

General Description

This program has been designed to help students read electrical meters correctly. Its vivid graphical display and lucid instructions mean that, after initial preparation, this program can be used unsupervised.

Detailed Description

Lines 100-250 Main Program.

270-460 PROCscreen: Selects colours, defines graphics and text windows using VDU24,28.

480-900 PROCscale: Draws meter and numbers scale using VDU29.

920-1070 PROCranges: Choices of f.s.d. and subdivisions is read into RANGE array from DATA lines 990-1020. Lines 1040 and 1050 randomly choose values of SF and SD%.

1090-1250 PROCpointer: Draws pointer. 1170 calculates number of subdivisions on chosen scale and 1180 converts this into angle of deviation.

1270-1520 PROCuser: This part of the program deals with computer-user interaction.

1540-1670 PROCtitle: Uses teletext Mode 7 to produce double height characters.

1690-1930 PROCinfo: Provides the program information.

1950-1990 This section generates error message, and offers options.

Educational Notes

In the classroom this program could be used in conjunction with a real meter to teach scale reading. It also lends itself to being used in a remedial situation with small groups of students who either missed or misunderstood the original lesson.

To modify the program to suit particular meters, just change the DATA values in PROCrange.

Program Listing

```

10 REM*****
20 REM**      THE      **
30 REM**      METER    **
40 REM**      READER   **
50 REM**                **
60 REM**      Alan Baugh **
70 REM**                **
80 REM**      Feb.1983  **
90 REM*****
100 MODE7
110 PROCtitle
120 PROCinfo
130 DIM RANGE(3,3)
140 SCORE=0
150 N=0
160 TRY=0
170 ON ERROR CLG:CLS:GOTO1950
180 MODE1
190 PROCscreen
200 PROCranges
210 PROCscale
220 PROCpointer
230 PROCuser
240 GOTO200
250 END
260 REM*****
270 DEFPROCscreen
280 REM*****
290 VDU19,2,6,0,0,0 :REM CYAN
300 VDU19,3,7,0,0,0 :REM WHITE
310 VDU19,1,4,0,0,0 :REM BLUE
320 COLOUR130:CLS
330 REM SET GRAPHICS WINDOW
340 VDU24,280;380;1000;920;
350 REM SET TEXT WINDOW
360 VDU28,0,31,39,22
370 COLOUR129
380 CLS
390 GCOL0,128
400 CLG
410 REM SHRINK GRAPHICS WINDOW
420 VDU24,290;390;990;910;
430 GCOL0,131
440 CLG
450 COLOUR3
460 ENDPROC
470 REM*****
480 DEFPROCscale:REM DRAWS SCALE
490 REM*****
500 REM DEFINE GRAPHICS ORIGIN
510 VDU29,640;380;
520 MOVE0,0

```

```

530 R%=50
540 GCOL0,0
550 FORA=0 TO PI STEP.2
560   X=R%*COS(A):Y=R%*SIN(A)
570   PLOT85,X,Y
580   PLOT85,0,0
590 NEXTA
600 R%=400
610 A=PI/4
620 X=R%*COS(A):Y=R%*SIN(A)
630 MOVEX,Y
640 FORA=PI/4 TO PI*.76 STEP.05
650   X=R%*COS(A):Y=R%*SIN(A)
660   PLOT5,X,Y
670 NEXTA
680 FOR S%=0 TO 5
690   A=PI*.75-(PI/2)/5)*S%
700   R%=400
710   X=R%*COS(A):Y=R%*SIN(A)
720   MOVEX,Y
730   R%=460
740   X=R%*COS(A):Y=R%*SIN(A)
750   PLOT5,X,Y:REM DRAW MAIN DIVISIONS
760   VDU5
770   MOVEX-300,Y+35
780   PRINTS%*SF/5 :REM PRINT NUMBERS
790   VDU4
800 NEXTS%
810 FORS%=0 TO 5*SD%
820   A=PI/4+(PI/2)/(5*D%)*S%
830   R%=400
840   X=R%*COS(A):Y=R%*SIN(A)
850   MOVEX,Y
860   R%=420
870   X=R%*COS(A):Y=R%*SIN(A)
880   PLOT5,X,Y:REM DRAW SUB DIVISIONS
890 NEXTS%
900 ENDPROC
910 REM*****
920 DEFPROCranges : REM SELECTS FSD AND SUB-DIVISIONS
930 REM*****
940 FOR CHOICE=0 TO 3
950   FOR SIZE=0 TO 3
960     READ RANGE(CHOICE,SIZE)
970     NEXT SIZE
980 NEXT CHOICE
990 DATA1,2,5,10
1000 DATA5,5,5,10
1010 DATA10,2,5,10
1020 DATA15,3,3,3
1030 N=RND(4)-1:M=RND(3)
1040 SF=RANGE(N,0):REM SELECTS F.S.D.
1050 SD%=RANGE(N,M):REM SELECTS SUB-DIVISIONS.
1060 RESTORE 990
1070 ENDPROC
1080 REM*****
1090 DEFPROCpointer
1100 REM*****
1110 MOVE15,0
1120 REM EASY START.
1130 IF SCORE<6 AND SD%=3 THEN 1160
1140 IF SCORE <6 AND SD%=5 THEN SD%=1
1150 IF SCORE <6 AND SD%=10 THEN SD%=2
1160 IF SCORE<3 THEN SD%=1
1170 INC=(PI/2)/(5*SD%)
1180 DEV=RND(5*SD%)*INC:REM DEVIATION
1190 A=PI*.75-DEV
1200 R=400
1210 X=R*COS(A):Y=R*SIN(A)

```

```

1220 DRAWX,Y
1230 PLOT85,-15,0
1240 VDU7
1250 ENDPROC
1260 REM*****
1270 DEFPROCuser
1280 REM*****
1290 ATTEMPT=0
1300 PRINTTAB(27,4)"YOUR SCORE"
1310 PRINTTAB(30,5)"IS:-"
1320 PRINTTAB(30,6)"----"
1330 PRINTTAB(31,7); SCORE
1340 PRINTTAB(30,8)"----"
1350 A=(DEV/(PI/2))*SF:REM CALCULATES CORRECT ANSWER 'A'.
1360 TRY=TRY+1
1370 ATTEMPT=ATTEMPT+1
1380 IF ATTEMPT > 3 THEN PRINTTAB(3,1)"THE CORRECT ANSWER I
S ";A:TIME=0:REPEAT:UNTIL TIME >400:CLG:PRINTTAB(3,1)SPC(115
):PRINTTAB(5,5)SPC(30):ENDPROC
1390 PRINTTAB(5,3)SPC(60)
1400 PRINTTAB(5,5)SPC(25)
1410 *FX15,1
1420 INPUTTAB(5,3)"THE READING IS =" B
1430 IF B>SF+1 THEN CLS:GOTO 1300
1440 B%=B*100
1450 A%=A*100
1460 IF (A%-B%)^2<3 THEN PRINTTAB(5,5);B;" IS RIGHT!":SCORE
=SCORE+1: ELSE PRINTTAB(5,5)"SORRY ";B;" IS WRONG":TIME=0:RE
PEAT:UNTILTIME>200: GOTO1300
1470 PRINTTAB(31,7); SCORE
1480 TIME=0:REPEAT:UNTIL TIME=200
1490 IF SCORE=12 THEN CLS:PRINTTAB(0,7)"YOU HAVE SCORED ";S
CORE;" POINTS FROM ";TRY;" TRIES"
1500 IF SCORE=12 THEN PRINTTAB(5,5)"DO YOU WANT TO START AG
AIN?":Z$=GET$:IF Z$="N"THEN CLS:PRINTTAB(15,5)"FAREWELL":END
ELSE CLS
1510 CLG
1520 ENDPROC
1530 REM*****
1540 DEFPROCtitle
1550 REM*****
1560 PRINTTAB(15,5)CHR$(141);CHR$(129);"THE"
1570 PRINTTAB(15,6)CHR$(141);CHR$(129);"THE"
1580 PRINTTAB(15,9)CHR$(141);CHR$(129);"METER"
1590 PRINTTAB(15,10)CHR$(141);CHR$(129);"METER"
1600 PRINTTAB(15,13)CHR$(141);CHR$(129);"READER"
1610 PRINTTAB(15,14)CHR$(141);CHR$(129);"READER"
1620 PRINTTAB(16,7)CHR$(131);"****"
1630 PRINTTAB(16,11)CHR$(131);"*****"
1640 PRINTTAB(16,15)CHR$(131);"*****"
1650 VDU30
1660 TIME=0:REPEAT:UNTIL TIME>400
1670 ENDPROC
1680 REM*****
1690 DEFPROCinfo
1700 REM*****
1710 CLS
1720 PRINTTAB(5,5)"The computer will draw the face"
1730 PRINT"of an electric meter with the pointer"
1740 PRINT"at a random position."
1750 PRINTTAB(5,9)"The full scale deflection will"
1760 PRINT"be";CHR$(131);"1, 5, 10";CHR$(135);"or";CHR$(131
);"15";CHR$(135);" units, divided into"
1770 PRINT"five sections. Each section will be"
1780 PRINT"divided into";CHR$(131);"2, 3, 5";CHR$(135);"or"
;CHR$(131);"10";CHR$(135);"sub-"
1790 PRINT"divisions."
1800 PRINTTAB(5,15)"You will be given three chances"
1810 PRINT"to read the meter. If you are still"

```

```

1820 PRINT"unsuccessful, you will be told the"
1830 PRINT"correct reading."
1840 PRINTTAB(0,23);CHR$(134);CHR$(136);"Press the SPACE  B
AR to continue"
1850 VDU30
1860 Z=GET
1870 CLS
1880 PRINTTAB(5,5)"After";CHR$(130);"twelve";CHR$(135);"cor
rect answers you"
1890 PRINT"will be told how well you have done."
1900 PRINTTAB(0,23);CHR$(134);CHR$(136);"Press the SPACE  B
AR to continue"
1910 VDU30
1920 Z=GET
1930 ENDPROC
1940 REM*****
1950 PRINTTAB(5,5)"There has been an error."
1960 PRINT
1970 PRINTSPC(5)"Do you want to start again?"
1980 Z$=GET$
1990 IFZ$="Y"THEN140
2000 REM*****

```

