
10. Rotation

General Description

This is an enormously frustrating game to be played against the
computer. The user must key in a two number vector which will
manifest itself as a line on the screen. Each of your moves will
be mirrored by the computer. Your objective is to locate the
centre and force the computer to meet your own line. The user
takes the blue line and any key can be pressed to start the
program. To run this program on the Model A, all the REM
statements should be removed.

Detailed Description

Lines 10-90 Credits.
100-160 Display title.
170-410 Fancy pattern (can be deleted if not enough

memory).
420-520 Display instructions.
530 Go to procedure to display grid.
540 Go to procedure to input move from keyboard.
550 Go to procedure to move computer.
560 Go to procedure to check if hit computer.
570 Go and do it again.
580-730 Make grid and display starting points.
740-860 Input vector from player and validate, if OK then

move player.
870-920 Procedure to move computer.
930-950 Procedure to check if computer is at the same point

as you are.
960-1040 Procedure to show score when finished.
1050-1140 Finish.

Educational Notes

This is essentially an intelligence test which appears to leave the
majority of adults floundering in grave despair. For some reason
(!) to most youngsters seem to get the hang of it fairly quickly.

Program Listing
>
   10 REM *****************************
   20 REM *         ROTATION          *
   30 REM *                           *
   40 REM *   WRITTEN FOR THE BBC-A   *
   50 REM *           BY              *
   60 REM *     ANDREW  PUSEY         *
   70 REM *        FEB  1983          *
   80 REM *****************************
   90 REM 
  100 MODE7 
  110 ENVELOPE 1,1,0,0,0,20,30,50,124,-2,-1,-1,126,126
  120 TITLE$="ROTATION"
  130 PRINT TAB(10,10);CHR$141;CHR$133;TITLE$
  140 PRINT TAB(10);CHR$141;CHR$133;TITLE$
  150 PRINT TAB(5,15);CHR$136;CHR$130;"Press any key to begi
n"
  160 A$=GET$
  170 MODE4
  180 FOR X=0TO1279 STEP 50
  190   MOVE X,0
  200   DRAW 640,1023
  210   MOVE X,1023
  220   DRAW 640,0
  230   MOVE0,X
  240   IFX<1024 DRAW1279,512
  250   MOVE 1279,X
  260   IF X<1024 DRAW 0,512
  270   SOUND 1,1,X/50,1
  280 NEXT
  290 A$=INKEY$(100)
  300 SOUND 1,1,100,10
  310 VDU19,1,5,0,0,0
  320 A$=INKEY$(70)
  330 SOUND 1,1,150,10
  340 VDU19,0,2,0,0,0
  350 A$=INKEY$(70)
  360 SOUND 1,1,200,10
  370 VDU 19,1,2,0,0,0
  380 VDU 19,0,0,0,0,0
  390 A$=INKEY$(80)
  400 SOUND 1,1,255,10
  410 MODE7
  420 PRINT TAB(2,4);"***** I N S T R U C T I O N S ******"
  430 PRINT TAB(1,7);CHR$129;"You input a vector to move."
  440 PRINT TAB(1,9);CHR$134;"The vector must be two numbers
."
  450 PRINT TAB(1,11);CHR$131;"The computer will reflect you
r move."
  460 PRINT TAB(1,13);CHR$130;"When you think you know where
 the"
  470 PRINT TAB(1,15);CHR$133;"center is, you must force the
"

61



  480 PRINT TAB(1,17);CHR$129;"computer's line to meet yours
."

  490 PRINT TAB(1,19);CHR$134;"You are the blue line"
  500 PRINT TAB(3,21);CHR$136;CHR$130;"Press any key to star
t"
  510 A$=GET$
  520 MODE 5:VDU23;8202;0;0;0;
  530 PROC_grid
  540 PROC_player
  550 PROC_computer
  560 PROC_check
  570 GOTO 540
  580 DEF PROC_grid
  590 GCOL0,1 
  600 FOR X=0 TO 1279 STEP 32
  610   MOVE X,0:DRAWX,1023
  620   MOVE 0,X:DRAW1279,X 
  630 NEXT
  640 CX=RND(20)*32+500:CY=RND(15)*32+400
  650 PX=RND(10)*32+500:PY=RND(10)*32+400
  660 VDU 19,2,5,0,0,0
  670 VDU 19,3,6,0,0,0
  680 MOV=0
  690 GCOL0,2 
  700 PLOT69,CX,CY
  710 GCOL0,3 
  720 PLOT69,PX,PY
  730 ENDPROC
  740 DEF PROC_player
  750 MOV=MOV+1
  760 PRINT TAB(0,0);SPC(20)
  770 PRINT TAB(0,0);:INPUT"Vector ",X,Y
  780 IF PX+X*32 < 0 OR PX+X*32 > 1279 THEN VDU7:GOTO760
  790 IF PY+Y*32 < 0 OR PY+Y*32 > 900 THEN VDU7:GOTO760
  800 IF CY-Y*32 < 0 OR CY-Y*32 > 900 THEN VDU7:GOTO760
  810 IF CX-X*32 < 0 OR CX-X*32 > 1279 THEN VDU7:GOTO760
  820 MOVE PX,PY
  830 PX=PX+X*32:PY=PY+Y*32
  840 GCOL0,3
  850 DRAW PX,PY
  860 ENDPROC
  870 DEF PROC_computer
  880 MOVE CX,CY
  890 CX=CX-X*32:CY=CY-Y*32
  900 GCOL0,2
  910 DRAW CX,CY
  920 ENDPROC
  930 DEF PROC_check
  940 IF CX=PX AND CY=PY PROC_finish
  950 ENDPROC
  960 DEF PROC_finish 
  970 A$=INKEY$(200)
  980 CLS
  990 COLOUR 3
 1000 PRINT'''"You took "TAB(10);MOV;" moves"
 1010 PRINT'''"Press SPACE to start again"
 1020 IF INKEY(-99)=FALSE THEN 1020
 1030 CLEAR
 1040 GOTO 410
 1050 DEFPROC_end
 1060 CLS
 1070 COLOUR3
 1080 PRINT TAB(4,4);"Continue (Y/N)";
 1090 A$=GET$
 1100 IFA$<>"Y" AND A$<>"N" THEN VDU7:GOTO1090
 1110 IF A$="Y"THEN ENDPROC
 1120 PRINT A$
 1130 PRINT '''"Thanks for playing"

 1140 END

