
27. Screen Wordgame

General Description

This is the traditional wordsquare puzzle played on the
computer. You are given half a dozen topics whose words you
may wish to have hidden in the wordsquare. You select the topic
and wait while the computer generates the wordsquare. When
the computer is ready it displays the wordsquare and you have to
move a cursor with the four cursor control keys around the
screen picking out the letters. When you think that you have
found a letter you may use the copy key to change its colour to
yellow. You will then notice that the letter is 'docked' from the
word displayed at the bottom which you had to find. Of course it
is possible to cheat and uncover every letter, in which case you
will inevitably score the points for the letters found, but I am
sure none of you would be that unsporting. This program runs on
a Model 'A' and you will notice that the final score is given in
terms of letters found rather than words. This is because it was
quite a squeeze to fit it into a Model 'A', and those with Model
'B's may wish to enhance the program. Let me know if you
succeed.

Detailed Description

See Printer Wordgame.

Educational Notes

The educational value of wordsquare hunts is dubious. It
familiarises youngsters with the words in the subject under study
but that is probably the limit. I have generated wordsquares for
180



other departments as requested and the other 'wordgame'
program by Ian Clarke generates your chosen words onto a
printer. I think that this program is more for use at home for
general educative entertainment. You are free, of course, to
change the data. Ian has chosen music as one of the wordsquare
subjects as it is one of his own interests.

Program Listing

   10  REM *****************************
   20  REM ** WORDGAME                **
   30  REM ** Written by Ian Clarke.  **
   40  REM ** Feb 83                  **
   50  REM *****************************
   60  ON ERROR GOTO 2220
   70  ENVELOPE 1,2,6,-6,0,8,8,1,1,-1,0,-1,120,100
   80  ENVELOPE 2,2,2,-4,4,2,3,1,0,-1,0,1,100,80
   90  ENVELOPE 3,2,4,5,6,10,6,8,1,-1,0,-1,120,100
  100  ENVELOPE 4,2,10,-4,15,20,20,5,2,0,0,-3,127,127
  110  CLEAR
  120  C1%=0 
  130  E%=0
  140  V%=0
  150  Y=0
  160  *FX4,1
  170  MODE 7
  180  REPEAT
  190    PRINT TAB(5,5+Y) CHR$(141) CHR$(131)"SQUAREWORDS"
  200    PRINT TAB(5,8+Y) CHR$(141) CHR$(131) "by Ian Clarke
"
  210    PRINT TAB(0,12+Y) CHR$(141) CHR$(133) "use the arro
w keys to move about the"
  220    PRINT TAB(0,14+Y) CHR$(141) CHR$(133) "cursor to th
e letter you want and hit"
  230    PRINT TAB(0,16+Y) CHR$(141) CHR$(133) "copy key if 
the letter is the one you"
  240    PRINT TAB(0,18+Y) CHR$(141) CHR$(133) "want to make
 a word with."
  250    PRINT TAB(0,20+Y) CHR$(141) CHR$(133) "Press ESCAPE
 to exit program"
  260    Y=Y+1
  270  UNTIL Y=2
  280  PRINT CHR$(131) "Type any key to continue":A$=GET$
  290  PROCarr
  300  OE$(1)="."
  310  OE$(2)="s."
  320  PROCinput
  330  PROCsqr
  340  PROCram
  350  GOTO 1500
  360  REM -----------------------------
  370  REM       Set up arrays.
  380  DEFPROCarr
  390  DIM C$(16,16),X%(8),Y%(8),T$(6),W$(10),OE$(2)
  400  X%(1)=1:X%(2)=1:X%(3)=0:X%(4)=-1
  410  X%(5)=-1:X%(6)=-1:X%(7)=0:X%(8)=1
  420  Y%(1)=0:Y%(2)=1:Y%(3)=1:Y%(4)=1
  430  Y%(5)=0:Y%(6)=-1:Y%(7)=-1:Y%(8)=-1
  440  REM .............................

  450  ENDPROC
  460  REM -----------------------------
  470  REM  Display the six topics and
  480  REM  input the user's choice.
  490  DEFPROCinput
  500  CLS
  510  SOUND 0,4,100,40
  520  FOR Y=2 TO 12 STEP 2
  530    READ T$(Y/2)
  540    PRINT TAB(16,Y);CHR$(141);CHR$(131);(Y)/2;" ";T$(Y/
2)
  550    PRINT TAB(16);CHR$(141);CHR$(131)(Y)/2;" ";T$(Y/2)
  560    FOR X=1 TO 500:NEXT

  570  NEXT Y
  580  PRINT TAB(0,8);CHR$(141);CHR$(133);"Which topic"
  590  PRINT TAB(0);CHR$(141);CHR$(133);"Which topic"
  600  PRINT TAB(0,10);CHR$(141);CHR$(133);"do you want?"
  610  PRINT TAB(0);CHR$(141);CHR$(133);"do you want?"
  620  *FX15

  630  D=GET
  640  IF D>48 AND D<55 GOTO 690
  650  VDU 7
  660  FOR X=1 TO 200:NEXT
  670  *FX15
  680  GOTO 630
  690  PRINT TAB(5);CHR$(141);CHR$(133);D-48
  700  PRINT TAB(5);CHR$(141);CHR$(133);D-48
  710  FOR X=1TO 1000:NEXT
  720  RESTORE (D-49)*30+1950
  730  FOR C=1 TO 10
  740    READ W$(C)
  750    C1%=C1%+LEN(W$(C))
  760  NEXT C
  770  DIM Y2%(C1%),X2%(C1%),T(C1%),S(C1%)
  780  V%=C1%
  790  C1=10
  800  REM .............................
  810  ENDPROC
  820  REM -----------------------------
  830  REM     Thinks of grid for
  840  REM     topic chosen.
  850  DEFPROCsqr
  860  CLS
  870  PRINT TAB(5,12);CHR$(141);CHR$(136);CHR$(133);"Thinki
ng of puzzle."
  880  PRINT TAB(5);CHR$(141);CHR$(136);CHR$(133);"Thinking 

of puzzle."
  890  FOR N%=1 TO 10
  900    W$=W$(N%)
  910    P%=RND(8)
  920    Y%=Y%(P%) : X%=X%(P%)
  930    X4%=RND(15):Y4%=RND(15)
  940    U$=C$(X4%,Y4%)
  950    IF U$="" THEN GOTO 1030
  960    I%=0
  970    REPEAT
  980      I%=I%+1
  990    IF MID$(W$,I%,1)=U$ THEN X4%=X4%-((I%-1)*X%):Y4%=Y4
%-((I%-1)*Y%):UNTIL I%>0:GOTO 1020
 1000  UNTIL I%=LEN(W$)
 1010  GOTO 910
 1020  IF X4%<1 OR X4%>15 OR Y4%<1 OR Y4%>15 THEN GOTO 910
 1030  X1%=X4%:Y1%=Y4%
 1040  W%=0
 1050  REPEAT
 1060    W%=W%+1

 1070  IF C$(X1%,Y1%)<>"" AND C$(X1%,Y1%)<>MID$(W$,W%,1) THE
N UNTIL W%>1:GOTO 910
 1080  X1%=X1%+X%:Y1%=Y1%+Y%
 1090  IF X1%<1 OR X1%>15 OR Y1%<1 OR Y1%>15 THEN UNTIL W%>1
:GOTO 900
 1100  UNTIL W%=LEN(W$)
 1110  X1%=X4%:Y1%=Y4%
 1120  FOR K%=1 TO LEN(W$)
 1130    C$(X1%,Y1%)=MID$(W$,K%,1)
 1140    E%=E%+1
 1150    Y2%(E%)=Y1%-1
 1160    X2%(E%)=X1%-1
 1170    X1%=X1%+X%:Y1%=Y1%+Y%
 1180    T(E%)=N%
 1190    S(E%)=K%
 1200  NEXT K%
 1210  NEXT N%
 1220  REM .............................
 1230  ENDPROC
 1240  DEFPROCram
 1250  REM -----------------------------
 1260  REM   Display grid on VDU.
 1270  CLS
 1280  FOR Y%=1 TO 15
 1290    FOR X%=1 TO 15
 1300      IF C$(X%,Y%)<>"" THEN GOTO 1320
 1310      C$(X%,Y%)=CHR$(RND(26)+64)
 1320      PRINT TAB((X%-1)*2,Y%-1);CHR$(130);C$(X%,Y%)
 1330    NEXT X%
 1340  NEXT Y%
 1350  Y=17
 1360  Q=0
 1370  FOR X=1 TO 10
 1380    IF Q=15 THEN Y=12
 1390    PRINT TAB(Q,X+Y);CHR$(133);W$(X)
 1400    IF X=5 THEN Q=15
 1410  NEXT
 1420  PRINT TAB(0,13)
 1430  VDU 9
 1440  X1=0:Y1=14
 1450  REM .............................
 1460  ENDPROC
 1470  REM   User enters cursor position
 1480  REM   or copy key.  The copy key
 1490  REM   enters letters.
 1500  FOR X=1 TO 200:NEXT
 1510  *FX15
 1520  C=GET
 1530  IF C>134 AND C<140 THEN GOTO 1560
 1540  VDU 7
 1550  GOTO 1500
 1560  ON C-134 GOSUB 1580,1780,1820,1860,1900
 1570  GOTO 1510
 1580  Q=0
 1590  S=0
 1600  REPEAT
 1610    Q=Q+1
 1620    IF Y2%(Q)=Y1 AND X2%(Q)=X1 THEN V%=V%-1:GOTO 1640
 1630  UNTIL Q=C1%:GOTO 1740
 1640  Y2%(Q)=-1
 1650  S=1
 1660  IF T(Q)>5 THEN X=15:Y=12 ELSE X=0:Y=17
 1670  PRINT TAB(X+S(Q),T(Q)+Y);" "
 1680  PRINT TAB(X1*2,Y1);CHR$(131);C$(X1+1,Y1+1)
 1690  FOR X=1 TO X1*2+1
 1700    VDU 9
 1710  NEXT X
 1720  VDU 11
 1730  UNTIL Q=C1%

 1740  IF S=1 THEN S=0:SOUND 1,2,120,5 ELSE SOUND 1,12,120,1
0
 1750  FOR X=1 TO 500:NEXT
 1760  IF V%=0 THEN GOTO 2140
 1770  RETURN
 1780  IF X1=0 THEN GOTO 1810
 1790  X1=X1-1
 1800  VDU 8,8
 1810  RETURN
 1820  IF X1=14 THEN GOTO 1850
 1830  X1=X1+1
 1840  VDU9,9
 1850  RETURN
 1860  IF Y1=14 THEN GOTO 1890
 1870  Y1=Y1+1
 1880  VDU 10
 1890  RETURN
 1900  IF Y1=0 THEN GOTO 1930
 1910  Y1=Y1-1
 1920  VDU 11
 1930  RETURN
 1940  DATA MUSIC,PHYSICS,ELECTRONICS,COMPUTER STUDIES,MATHS
,SPORT
 1950  DATA ORGAN,FRENCHHORN,BEETHOVEN,STRAVINSKY
 1960  DATA MUSICSTAND,CLARINET,CORANGLAIS
 1970  DATA RECORDER,TRUMPET,CONCERTHALL
 1980  DATA ELECTRIC,NEUTRON,NEWTON,DENSITY
 1990  DATA HOOKESLAW,ELASTICITY,PHYSICS
 2000  DATA MOMENTS,REFRACTION,LENSES
 2010  DATA OSCILLOSCOPE,TRANSISTOR,VEROBOARD,CATHODE
 2020  DATA INTEGRATED,TRANSDUCER,FEEDBACK
 2030  DATA CAPACITOR,CIRCUIT,RESISTOR
 2040  DATA MICROCOMPUTER,ASSEMBLER,COMPILERR
 2050  DATA INTERPRETER,DISKETTES,PROCESSOR
 2060  DATA MICROCHIP,DISKDRIVE,MONITOR,KEYBOARD           
 2070  DATA DIVISION,TRIGONOMETRY,COSINERULE,MULTIPLICATION
 2080  DATA ADDITION,SINE,PERCENTAGE
 2090  DATA INTEGRATION,MATRIX,GEOMETRY
 2100  DATA FOOTBALL,SQUASH,CURLING,POLEVAULTING
 2110  DATA STEEPLECHASE,GYMNASTICS,TABOGANNING
 2120  DATA ICEHOCKEY,BADMINTON,HORSERACING
 2130  REM Print congratulation message.
 2140  CLS
 2150  SOUND 1,3,100,50
 2160  PRINT TAB(3,8);CHR$(141);CHR$(129);"Well done.  You f
ound all"
 2170  PRINT TAB(3);CHR$(141);CHR$(129);"Well done.  You fou
nd all"
 2180  PRINT TAB(11);CHR$(141);CHR$(129);"the words."
 2190  PRINT TAB(11);CHR$(141);CHR$(129);"the words."
 2200  FOR X=1 TO 4000:NEXT
 2210  GOTO 2270
 2220  CLS
 2230  IF C1%-V%=0 THEN GOTO 2290
 2240  IF C1%-V%=1 THEN Q=1:ELSE Q=2
 2250  PRINT TAB(8,10);CHR$(141);CHR$(129);"You found "C1%-V
%;" letter";OE$(Q)
 2260  PRINT TAB(8);CHR$(141);CHR$(129);"You found "C1%-V%;"
 letter";OE$(Q)
 2270  IF C1%-V%=1 THEN Q=1 ELSE Q=2
 2280  FOR X=1 TO 4000:NEXT
 2290  PRINT TAB(5,12);CHR$(141);CHR$(129);"Do you want anot
her game?"
 2300  PRINT TAB(5);CHR$(141);CHR$(129);"Do you want another
 game?"
 2310  *FX15
 2320  YN$=GET$
 2330  IF YN$="Y" OR YN$="y" THEN GOTO 110
 2340  SOUND 1,1,90,70

 2350  PRINT TAB(14);CHR$(141);CHR$(136);CHR$(129);"BYE!!!"
 2360  PRINT TAB(14);CHR$(141);CHR$(136);CHR$(129);"BYE!!!"
 2370  FOR X=1 TO 10000:NEXT
 2380  CLS
 2390  *FX4
 2400  END

