

21. Attributes

General Description

This program tries to provide a forum for you to develop question and answer programs for the classroom. Two sets of example data are provided. The science data was developed for less able fourth years in a girls' school. The geography data is included to show the program's adaptability.

The program generates a strange introduction sound leading to an introduction page on which the scoring and rules are displayed. You start with 50 points and lose a small number of points for the help that you require - the largest loss being 5 points for a fruitless guess. Entry is single key, with a confirm check for every data entry except the final guess, to which you are committed. Success is rewarded with a 50 point bonus and a chance to play the game with your new total.

The main program displays a screen of up to eight YES/NO questions. Keying the number next to the question generates a response to the question. Typing 'H' moves onto the HELP pages. A list of all the possible answers is free and is obtained by typing 'G' for 'general' help. Special hints are obtained by typing 'S', 'R' returns you to the main program. In the main program 'Z' generates the surrender page and noises, while 'X' allows you to enter your guess.

It is possible but very difficult to use up all your 50 starting points, in which case you are deemed to have failed and a suitable message is created. When you lose all your points or 'give-up' the correct answer is displayed along with all the correct YES/NO attributes.

ADAPTABILITY

For those who wish to develop their own data, the following points should be noted.

Line 4580 contains the title (30 characters or less).

4590 contains the repetitive part of the free hint message. If nothing is to be indicated here then enter "" as the data.

4600 contains the number of the questions to be asked. This may vary from one to eight. Eight fits nicely onto the screen but the screen does readjust for other totals. Greater than eight questions, means that you will have to rewrite the screen display and update procedures to fit the extra questions in.

4610 contains the number of data items that follow. This **MUST** be an **EVEN** number of data items. The program has been tested up to 34 data items, and should run happily for up to a hundred, but if you choose a larger number of data items, you will find the **HELP** pages take a little longer to come up.

4620 up to 5990 may contain data. This must be in the following format: the item being tested, followed by its code (see below), followed by the free hint. If you are not having a free hint, this must be a "".

DATA CODES

The clue to these codes is in data line 4520. Each question has a number equivalent to the question. Thus question one has value 1 while question eight has value 128. ' PARIS has code 1 because ' its a capital' (question 1) and code 4 because ' its on a river' (question three). Add these together and the code for ' PARIS is ... QED.

IF YOU DO NOT HAVE EIGHT QUESTIONS

The validation checks will need to be changed in lines 1600 and line 1630 if you decide not to have eight questions. Thus for four questions only, line 1600 would read: 1600 Check\$ = ' 1234ZXH' :GOTO1650.

DO NOT RENUMBER THE PROGRAM or the computed

RESTORE in lines 1980 and 1990 will fail.

QUESTIONS

These must occupy data lines 6000 and upwards.

For a program which is 500 statements long, there are remarkably few changes that need to be made to make the program run for a completely different subject area.

Detailed Description

Lines 10-370 The main structure: this is virtually self-documenting. It calls sub-structures ' help' and ' question' . QUESTION in line 230 is set to zero by default if a question number is not pressed and Z, X or H are pressed.

380-670 These are two main sub-structures. The flag SPECIAL controls which advice will be displayed in line 2630.

680-1020 It is possible but inadvisable to radically alter the screen display. I have fiddled with it to get it as it is. However note that the ' 9 in line 1000 is one less than the number of non-question display lines on the screen, starting from the top. There are several different methods of doing line 1010 - try for yourself and tell me.

1030-1120 Note that the question ' getarea is cleared at lines 1110 and 1120.

1230-1500 The guts of the program lies in the logical ' AND' performed in line 1330. For the technical, the bit-value is its decimal value flagged on an 8 bit register. ' Value is obtained from each items' data line. The bit values are given in line 4520.

1510-1860 The validation is based on a suggestion I give elsewhere in the book. The delay routine is as suggested in the manual.

1870-2420 Don' t fiddle with the initialisation unless you know what you are doing. The screen introduction you can improve upon, I suspect.

2430-2630 Unlike the rest of the help menus, these are easy to follow.

2660-3520 Though the special and general help routines have

much in common, there is not sufficient to warrant drawing on similar procedures. Lines 2810-2850 determine the fact that there must be even numbers of data. Line 3400 performs another logical AND to sort out the final display for losers. If you can read the indenting on BBC Basic then the REPEATS fall into place - suffice it to say that exit back to the main program involves falling out of the layer of the onion your loop has currently crept into.

3530-4370 These may be altered at will. The A% in lines 4770 and 4290 are essential as the CLEAR in line 160 will clear all but the ' resident variables A% - Z%".

4380-4480 Keep these error trapping lines - you never know what you will find.

Data lines explained above.

Educational Notes

This program should be used in small groups, a maximum of four, round a single machine. The structure of the program lends itself to the less able, unless for the more able you are prepared to generate a large and challenging amount of data. The only point that the users of the program must realise, is to use the general help page for the set of answers. Looking at the science data, you can understand why: a sulphur rod and sulphur powder have some different physical characteristics and so sulphur on its own cannot be a suitable response. Similarly the ruler must be wooden!

Program Listing

```

10 REM =====
20 REM     Attributes by Siu Ming
30 REM     and Ian Murray for the
40 REM     BBC 'A' & 'B' from an
50 REM     idea by Siu Ming.
60 REM     Written January 1983
70 REM     =====
80 REM
90 REM     =====
100 REM          Main Structure
110 REM     .....
120 REM
130 *KEY10 OLD|MRUN|M
140 ON ERROR GOTO 4380
150 A%=50

```

```

160 CLEAR : SCORE = A%
170 CLS
180 PROCinitialise
190 PROCintroduce
200 PROCquestion_display
210 REPEAT
220     REPEAT
230         QUESTION = 0
240         PROCget_question
250         IF QUESTION > 0 THEN PROCquestion
260         UNTIL GUESS OR HELP OR GIVEUP OR FAILED
270         CORRECT = FALSE
280         IF HELP THEN PROCChelp
290         IF GUESS THEN PROCguess
300     UNTIL FAILED OR GIVEUP OR CORRECT
310     IF FAILED PROCToo_long
320     IF GIVEUP PROCsurrender
330     IF CORRECT PROCwin ELSE PROCclose
340 PROCfinal
350 IF YES GOTO 160
360 PRINT TAB(5) "BYE"
370 END
380 REM =====
390 REM         help
400 REM .....
410 REM
420 DEF PROCChelp
430 SATISFIED = FALSE
440 REPEAT
450     REPEAT
460         CLS
470         PROCChelp_menu
480         SPECIAL = FALSE
490         PROCvalidate(3)
500         PRINT CHR$(134) "Confirm your selection (Y/N) ";
510         PROCvalidate(2)
520     UNTIL YES
530     IF NOT SATISFIED THEN PROCdisplay_advice
540     CORRECT = FALSE
550     PROCquestion_display
560     PROCscore
570     ENDPROC
580 REM =====
590 REM     question?
600 REM .....
610 REM
620 DEF PROCquestion
630 PROCanalyse
640 PROCscore
650 PROCupdate_screen(QUESTION)
660 ENDPROC
670 REM =====
680 REM         screen display
690 REM .....
700 REM
710 DEF PROCquestion_display
720 LOCAL X,Y
730 X=0:CLS
740 Y = (40-LEN(Title$))/2
750 REPEAT
760     PRINT TAB(Y) CHR$(131) CHR$(141) Title$
770     X = X+1
780     UNTIL X=2
790     PRINT CHR$(134) STRING$(38,"-")
800     PRINT CHR$(130) "HINT "CHR$(135) Hint$
810     PRINT CHR$(134) STRING$(38,"-")
820     PRINT CHR$(133) "Type question number to ask questi
830 on"
840     PRINT TAB(8) CHR$(131) "X = to make a guess"

```

```

850 PRINT TAB(8) CHR$(131) "Z = to give up"
860 PRINT TAB(8) CHR$(131) "H = to ask for help"
870 PRINT CHR$(134) STRING$(38,"-")
880 FOR X = 1 TO ATTRIBUTES
890 PRINT;" " X " " CHR$(134) Question$(X) TAB(35) Cl
ue$(X)
900 NEXT
910 PRINT CHR$(134) STRING$(38,"-")
920 PRINT;"Your score is: "SCORE
930 PRINT CHR$(134) STRING$(17,"-")
940 ENDPROC
950 REM =====
960 REM screen updating
970 REM .....
980 REM
990 DEF PROCupdate_screen(X)
1000 PRINT TAB(35,9+X) Clue$(X)
1010 IF RIGHT$(Clue$(X),1)="S" THEN SOUND 1,-10,180,4 EL
SE SOUND 1,-10,40,4
1020 ENDPROC
1030 REM =====
1040 REM get the selection
1050 REM .....
1060 REM
1070 DEF PROCget_question
1080 REPLY$ = ""
1090 REPEAT
1100 HELP = FALSE:GIVEUP = FALSE:GUESS = FALSE
1110 VDU 31,0,21
1120 PRINT SPC(38) " ":PRINT SPC(38) " "
1130 VDU 31,0,21
1140 PRINT "What do you want ? ";
1150 PROCvalidate(1)
1160 PRINT "Confirm your selection (Y/N) ";
1170 PROCvalidate(2)
1180 UNTIL YES
1190 IF REPLY$ = "H" THEN HELP = TRUE
1200 IF REPLY$ = "Z" THEN GIVEUP = TRUE
1210 IF REPLY$ = "X" THEN GUESS = TRUE
1220 ENDPROC
1230 REM =====
1240 REM analysis and results
1250 REM .....
1260 REM
1270 DEF PROCanalyse
1280 RESTORE 4520
1290 LOCAL X
1300 FOR X=1 TO QUESTION
1310 READ Bit_value
1320 NEXT
1330 IF Value AND Bit_value THEN Clue$(QUESTION) = CHR$(
131)+ "YES" ELSE Clue$(QUESTION) = CHR$(133)+ "NO "
1340 ENDPROC
1350 REM .....
1360 DEF PROCguess
1370 VDU 31,0,21
1380 PRINT SPC(38) " ":PRINT SPC(38) " "
1390 VDU 31,0,21
1400 INPUT"Enter your guess > "Guess$
1410 SOUND 1,-10,75,4
1420 IF Guess$ = WORD$ THEN CORRECT = TRUE ELSE PRINT CH
R$(134) " < WRONG > lose 5 marks":PROCdelay(2):PROCscore
1430 ENDPROC
1440 REM .....
1450 DEF PROCscore
1460 IF GUESS AND NOT CORRECT THEN SCORE = SCORE - 5
1470 IF NOT GUESS AND NOT HELP THEN SCORE = SCORE - 1
1480 IF SCORE < 1 THEN FAILED = TRUE
1490 PRINT; TAB(16,11+ATTRIBUTES) SCORE " "

```

```

1500 ENDPROC
1510 REM =====
1520 REM         validate
1530 REM .....
1540 REM
1550 DEF PROCvalidate(Type)
1560 LOCAL X,Check$,B$
1570 X=0:B$=""
1580 YES=FALSE
1590 ON Type GOTO 1600,1610,1620,1630
1600 Check$ = "12345678ZXH":GOTO 1650
1610 Check$ = "yYnN":GOTO 1650
1620 Check$ = "RGS":GOTO 1650
1630 Check$ = "12345678"
1640 *FX15,0
1650 REPEAT
1660     B$=GET$
1670     X=INSTR(Check$,B$)
1680     SOUND 1,-10,150,4
1690     PRINT B$
1700     IF Type = 1 AND X > 8 THEN REPLY$ = B$
1710     IF Type = 1 AND X < 9 THEN QUESTION = VAL(B$)
1720     IF Type = 4 THEN QUESTION = VAL(B$)
1730     IF Type = 2 AND X < 3 THEN YES = TRUE
1740     IF Type = 3 AND B$="S" THEN SPECIAL = TRUE
1750     IF Type = 3 AND B$="R" THEN SATISFIED = TRUE
1760     IF Type = 3 AND B$="R" THEN SATISFIED = TRUE
1770 ENDPROC
1780 REM =====
1790 REM         delay
1800 REM .....
1810 REM
1820 DEF PROCdelay(DEL)
1830 FINISH = TIME + 100*DEL
1840 REPEAT
1850 UNTIL TIME > FINISH
1860 ENDPROC
1870 REM =====
1880 REM     initialisation
1890 REM .....
1900 REM
1910 DEF PROCinitialise
1920 LOCAL START
1930 DIM Question$(8),Clue$(8)
1940 FAILED = FALSE:CORRECT = TRUE
1950 ENVELOPE 1,135,2,1,-5,40,35,20,5,0,0,5,180,80
1960 RESTORE 4580
1970 READ Title$,Message$,ATTRIBUTES,HOW_MANY
1980 START = 4610 + (RND(HOW_MANY)*10)
1990 RESTORE START
2000 READ WORD$,Value,Hint$
2010 Hint$ = Message$ + " " + Hint$
2020 RESTORE 6000
2030 FOR X = 1 TO ATTRIBUTES
2040     READ Question$(X)
2050     Clue$(X) = " ??? "
2060 NEXT
2070 ENDPROC
2080 REM =====
2090 REM     introduction
2100 REM .....
2110 REM
2120 DEF PROCintroduce
2130 SOUND 2,1,80,140
2140 LOCAL X
2150 VDU 31,0,11
2160 REPEAT
2170     PRINT TAB(5) CHR$(134) CHR$(141) Title$
2180     X=X+1
2190 UNTIL X=2

```

```

2200      PROCdelay(5)
2210      CLS
2220      PRINT CHR$(131) "This program sees if you can wor
k out"
2230      PRINT CHR$(131) "what a thing is from just a very
few"
2240      PRINT CHR$(131) "clues."
2250      PRINT
2260      PRINT CHR$(131) "You are charged 1 point for each
"
2270      PRINT CHR$(131) "question you choose to ask."
2280      PRINT CHR$(131) "You are charged 5 points for a g
uess!"
2290      PRINT CHR$(131) "You are charged 3 points for spe
cial"
2300      PRINT CHR$(131) "help. General help is FREE"
2310      PRINT
2320      PRINT CHR$(131) "Type" CHR$(133)"H" CHR$(131) "fo
r special and general"
2330      PRINT CHR$(131) "help. This will give a list of a
ll the"
2340      PRINT CHR$(131) "possible answers!"
2350      PRINT CHR$(131) "A GUESS MUST BE SPELT CORRECTLY"
2360      PRINT:PRINT
2370      PRINT CHR$(134) "REMEMBER TO USE" CHR$(131) "HELP
" CHR$(134) "BEFORE A GUESS"
2380      PRINT
2390      PRINT "Type any key to go on.";
2400      GO$ = GET$
2410      SOUND 1,-10,180,4
2420      ENDPROC
2430      REM =====
2440      REM      help menus etc.
2450      REM .....
2460      REM
2470      DEF PROChelp_menu
2480      CLS
2490      PRINTTAB(18) CHR$(130) CHR$(141) "HELP"
2500      PRINTTAB(18) CHR$(130) CHR$(141) "HELP"
2510      PRINT:PRINT
2520      PRINT CHR$(131) "For special help type 'S'"
2530      PRINT CHR$(131) "For general help type 'G'"
2540      PRINT CHR$(131) "To return to game type 'R'"
2550      PRINT
2560      PRINT CHR$(131) "Remember you are charged 3 point
s for"
2570      PRINT CHR$(131) "each special help you receive."
2580      PRINT
2590      PRINT CHR$(134) "Enter what you want to do > ";
2600      ENDPROC
2610      REM .....
2620      DEF PROCdisplay_advice
2630      IF SPECIAL THEN PROCspecial ELSE PROCgeneral
2640      ENDPROC
2650      REM .....
2660      DEF PROCgeneral
2670      LOCAL X,Y,Z,Z$,Q
2680      RESTORE 4620
2690      REPEAT
2695      Z=0
2700      REPEAT
2710      X=0:Y=0
2720      CLS
2730      REPEAT
2740      PRINT TAB(15) CHR$(141) CHR$(131) "GENERAL
HELP"
2750      X=X+1
2760      UNTIL X=2
2770      PRINT:PRINT CHR$(133) "Here are the possible

```



```

answers."
2780          VDU 31,0,5
2790          REPEAT
2800              X=0
2810              REPEAT
2820                  READ HELP$,Q,Z$
2830                  PRINT TAB(5+20*X) HELP$;
2840                  X=X+1:Y=Y+1:Z=Z+1
2850              UNTIL X=2
2860              PRINT
2870              IF Y=HOW_MANY THEN RESTORE 4620
2880              UNTIL Y=30 OR Z=HOW_MANY
2890              PRINT
2895              *FX15,0
2900              IF Y=30 THEN PRINT CHR$(131) "Type a key to c
ontinue";:Z$=GET$:SOUND 1,-10,180,4
2910              UNTIL Z=HOW_MANY
2920              PRINT
2930              PRINT CHR$(131) "Have you finished with the hel
p (Y/N) ";
2940                  PROCvalidate(2)
2950                  UNTIL YES
2960                  ENDPROC
2970                  REM .....
2980                  DEF PROCspecial
2990                  LOCAL W,X,Y,Z,Q,Z$
3000                  REPEAT
3010                      RESTORE 4520
3030                      X=0:Z=0:W=0
3040                      CLS
3050                      REPEAT
3060                          PRINT TAB(15) CHR$(141) CHR$(131) "SPECIAL HE
LP"
3070                          X=X+1
3080                          UNTIL X=2
3090                          PRINT:PRINT CHR$(133) "Which question do you wa
nt help for?"
3100                          PRINT
3110                          FOR X = 1 TO ATTRIBUTES
3120                              PRINT;" "X CHR$(131) Question$(X)
3130                          NEXT
3140                          PRINT
3150                          REPEAT
3160                              VDU 31,0,18
3170                              PRINT SPC(38) " ":PRINT SPC(38) " "
3180                              VDU 31,0,18
3190                              PRINT CHR$(133) "Enter the question number >
";
3200                              PROCvalidate(4)
3210                              PRINT CHR$(133) "Confirm your selection (Y/N)
";
3220                              PROCvalidate(2)
3230                              UNTIL YES
3240                              FOR X=1 TO QUESTION
3250                                  READ W
3260                              NEXT
3270                              RESTORE 4620
3275                              REPEAT
3280                                  CLS
3290                                  X=0:Y=0
3300                                  REPEAT
3310                                      PRINT; TAB(5) CHR$(134) CHR$(141) "HINTS FO
R QUESTION "QUESTION
3320                                      X=X+1
3330                                      UNTIL X=2
3340                                      PRINT:PRINT CHR$(133) Question$(QUESTION)
3350                                      PRINT CHR$(131) "For all these below the answ
er is YES"
3360                                      VDU 31,0,6

```

```

3370      REPEAT
3390          READ HELP$,Q,Z$
3400          IF W AND Q THEN PRINT TAB(5) HELP$:Y=Y+1
3410          Z=Z+1
3420          IF Y= HOW_MANY THEN RESTORE 4620
3430          UNTIL Y=15 OR Z=HOW_MANY
3440          PRINT
3445          *FX15,0
3450          IF Y=15 THEN PRINT CHR$(131) "Type a key to c
ontinue";:Z$=GET$:SOUND 1,-10,180,4
3460          UNTIL Z=HOW_MANY
3470          PRINT
3480          PRINT CHR$(131) "Have you finished with the hel
p (Y/N) ";
3490          PROCvalidate(2)
3500          SCORE = SCORE - 3
3510          UNTIL YES
3520          ENDPROC
3530          REM =====
3540          REM      end game routines
3550          REM .....
3560          REM
3570          DEF PROCToo_long
3580          LOCAL X
3590          CLS
3600          SOUND 0,-15,2,65
3610          X=0
3620          REPEAT
3630              VDU 31,9,3+X
3640              PRINT CHR$(134) CHR$(141) "YOU"
3650              VDU 31,12,6+X
3660              PRINT CHR$(134) CHR$(141) "HAVE"
3670              VDU 31,18,9+X
3680              PRINT CHR$(134) CHR$(141) "RUN"
3690              VDU 31,18,12+X
3700              PRINT CHR$(134) CHR$(141) "OUT"
3710              VDU 31,21,15+X
3720              PRINT CHR$(134) CHR$(141) "OF"
3730              VDU 31,24,18+Y
3740              PRINT CHR$(134) CHR$(141) "POINTS"
3750              X=X+1
3760          UNTIL X=2
3770          PROCdelay(3)
3780          ENDPROC
3790          REM .....
3800          DEF PROCsurrender
3810          LOCAL X
3820          CLS
3830          SOUND 0,-15,1,65
3840          PRINT TAB(14,11) CHR$(134) CHR$(141) CHR$(136) "C
HICKEN"
3850          PRINT TAB(14) CHR$(134) CHR$(141) CHR$(136) "CHIC
KEN"
3860          PROCdelay(3)
3870          ENDPROC
3880          REM .....
3890          DEF PROCwin
3900          CLS
3910          ENVELOPE 2,134,-5,-5,5,30,20,100,5,5,5,5,150,150
3920          SOUND 2,2,200,180
3930          LOCAL X
3940          X=0
3950          REPEAT
3960              VDU 31,11,8+X
3970              PRINT CHR$(133) CHR$(141) CHR$(136) "CONGRATULA
TIONS"
3980              VDU 31,15,12+X
3990              PRINT CHR$(131) CHR$(141) "YOU ARE"
4000              VDU 31,15,14+X

```

```

4010         PRINT CHR$(131) CHR$(141) "CORRECT"
4020         X=X+1
4030     UNTIL X=2
4040     VDU 31,7,18
4050     PRINT "If you choose to play again, you"
4060     PRINT TAB(7) "start with 50 extra points !!"
4070     A% = SCORE + 50
4080     PROCdelay(5)
4090     ENDPROC
4100     REM .....
4110     DEF PROCclose
4120     LOCAL X,Y
4130     CLS
4140     PRINT TAB(10) CHR$(130) CHR$(141) "SO YOU HAVE LO
ST"
4150     PRINT TAB(10) CHR$(130) CHR$(141) "SO YOU HAVE LO
ST"
4160     PRINT:PRINT
4170     X=0:Y=0
4180     FOR X = 1 TO ATTRIBUTES
4190         PRINT; " "X" " CHR$(131) Question$(X);
4200         QUESTION = X
4210         PROCanalyse
4220         PRINT TAB(35) Clue$(X)
4230     NEXT
4240     PRINT:PRINT
4250     PRINT CHR$(130) "The answer is : " CHR$(135) CHR$(
136) WORDS$
4260     PRINT TAB(5,21) CHR$(131) "TYPE ANY KEY TO GO ON
";
4270     *FX 15,0
4280     GO$=GET$
4290     A% = 50
4300     ENDPROC
4310     REM .....
4320     DEF PROCfinal
4330     CLS
4340     PRINT TAB(3,11) CHR$(134) CHR$(136) "Do you want
to play again (Y/N) ";
4350     PROCvalidate(2)
4360     PROCdelay(1.5)
4370     ENDPROC
4380     REM =====
4390     REM      whos a silly boy
4400     REM .....
4410     REM
4420     IF ERR = 17 GOTO 160
4430     CLS
4440     PRINT TAB(5) "PROGRAMMER ERROR !!"
4450     REPORT:PRINT; " at line "ERL
4460     PRINT:PRINT "Please document sequence to error"
4470     PRINT CHR$(131) "MY APOLOGIES"
4480     END
4490     REM =====
4500     REM      data
4510     REM .....
4520     DATA 1,2,4,8,16,32,64,128
4530     REM .....
4540     REM      the following data may
4550     REM      be changed provided you
4560     REM      keep the general structure
4570     REM .....
4580     DATA WORLD CITIES
4590     DATA The continent is
4600     DATA 8
4610     DATA 24
4620     DATA PARIS,5,EUROPE
4630     DATA HONG KONG,219,ASIA
4640     DATA CANBERRA,185,AUSTRALASIA

```

4650 DATA LONDON,53,EUROPE
 4660 DATA MOSCOW,5,EUROPE
 4670 DATA LOS ANGELES,18,NORTH AMERICA
 4680 DATA CALCUTTA,142,ASIA
 4690 DATA RIO DE JANEIRO,6,SOUTH AMERICA
 4700 DATA CANTON,12,ASIA
 4710 DATA TEHERAN,9,ASIA
 4720 DATA LUSAKA,137,AFRICA
 4730 DATA WEST BERLIN,4,EUROPE
 4740 DATA FLORENCE,4,EUROPE
 4750 DATA HO CHI MIN CITY,12,ASIA
 4760 DATA CHICAGO,148,NORTH AMERICA
 4770 DATA NEW YORK,150,NORTH AMERICA
 4780 DATA BUDAPEST,5,EUROPE
 4790 DATA CRAKOW,4,EUROPE
 4800 DATA CAIRO,15,AFRICA
 4810 DATA KINGSTON,187,CENTRAL AMERICA
 4820 DATA STOCKHOLM,37,EUROPE
 4830 DATA COLOGNE,4,EUROPE
 4840 DATA MONTE CARLO,99,EUROPE
 4850 DATA THE VATICAN,65,EUROPE
 6000 DATA Is it a captial city
 6010 DATA Is it on a coastline
 6020 DATA Is it on a river
 6030 DATA Is it a hot country
 6040 DATA Do they speak English there
 6050 DATA Does the country have a monarch
 6060 DATA Is it a city state
 6070 DATA Was it a British Colony

4580 DATA SCIENCE TESTER
 4590 DATA Its colour is
 4600 DATA 8

S

4610 DATA 20
 data
 4620 DATA WOODEN RULER,169,BROWN
 4630 DATA SAND,0,YELLOW
 4640 DATA SALT,2,WHITE
 4650 DATA SUGAR,98,WHITE
 4660 DATA POLYTHENE SHEET,225,TRANSPARENT
 4670 DATA IRON BAR,20,GREY
 4680 DATA PLASTICINE,224,ANY COLOUR
 4690 DATA GLASS ROD,200,TRANSPARENT
 4700 DATA PERSPEX VISOR,233,TRANSPARENT
 4710 DATA GOLD COIN,132,YELLOW
 4720 DATA GRAPHITE ROD,172,BLACK
 4730 DATA BRASS BAR,20,YELLOW
 4740 DATA SULPHUR ROD,232,YELLOW
 4750 DATA DIAMOND,32,TRANSPARENT
 4760 DATA MARBLE,152,ANY COLOUR
 4770 DATA CANDLEWAX,233,ANY COLOUR
 4780 DATA LEAD PIPE,148,GREY
 4790 DATA CHALK POWDER,16,WHITE
 4800 DATA ICE BLOCK,203,TRANSPARENT
 4810 DATA PAVING SLAB,136,GREY
 6000 DATA Does it float in water
 6010 DATA Does it dissolve in water
 6020 DATA Will it conduct electricity
 6030 DATA Does it snap if bent
 6040 DATA Will it react with acid
 6050 DATA Can a bunsen set it alight
 6060 DATA Will it melt in a bunsen
 6070 DATA Is it easy to scratch

