
14. N-Particle

General Description

The gravitational attraction between any two masses is given by
the formula:

F = G * m1 * m2
D*D

In the program the forces on each mass due to all the other
masses are added up so that its new position after time 't' may be
estimated. To minimise cumulative errors, all masses are moved
at the same time - i.e. all the forces on all the masses and all their
positions (calculated using the old positions), are calculated
before they are all moved. (Thus, at least:

Force of Mass A on Mass B = Force Mass B on Mass A
i.e. at least the momentum is conserved even for long sampling
times.)

A further refinement in the program is that rather than using
the instantaneous acceleration (which will always be too little
when masses are approaching and too large when they are
separating), the mean of the two latest and the previous
acceleration is used (except, of course, for the first calculation).

Detailed Description

The main program consists of three nested loops.
The innermost loop takes the mass at a time (Ms, for

example) and its data, and goes round the loop (N - 1) times to
calculate the total forces exerted on Ms by the other (N - 1)
masses. (This loop having, of course, the 'OLD' positions of all
the masses.)

The next loop out feeds masses one at a time to the inner
loop, and when each emerges with its X & Y acceleration
calculates its 'NEW' velocity (P & Q) components and
coordinates (stored in ALSO(S, O) and ALSO(S, 1)).

When all the masses have been processed, the outermost loop
uses a mini-loop (360-370) to update all the information used by
the other loops, and returns to the start.

Lines 10-120 Descriptions, instructions and set up arrays etc.
130-200 Accepts DATA and stores it in DAT(10, 5) (Notice

that the sampling time may be changed if greater accuracy is
required).

210 Outer Loop.
220 S=0 This resets the count for the centre loop - i.e. starts it

all again with the new data (updated in the lower half of the
loop).

230 Centre Loop (to 350).
The top half of this loop takes one mass at a time, (S=0 to n)Ms,
for example, gives the variables M, X, Y, U and V its coordinates
and velocity from SAT, then feeds those variables to the inner
loop (it also resets the count with H=0).

250 Inner Loop.
This inner loop finds the forces due to the H mass on the mass
Ms for all the masses - i.e. each time round it calculates the force
on Ms due to another mass and adds it to the previous total.

290 B & C hold the running total of the X & Y acceleration.
320 ACC holds accelerations. The old values are transformed

to E & F, before ACC is updated (with B & C). B & C are then
recalculated using B & E or C & F (to get the average over the
time interval 't').

K is 0 the first time round and thereafter 0.5.
330 P & Q are the new components of velocity calculated

using
v = u + ft

They are then stored directly in DAT. The new positions
calculated using

s = ut + ½ft ^ 2
are stored in the array ALSO until all particles have been treated.

360-380 Mini-loop updates all masses by transferring ALSO
to DAT.

Educational Notes

In the classroom this program will prove invaluable for the
opportunities it offers teachers to provide students with a lucid
demonstration of the laws of gravity. If restricted to two masses
with the initial momentum at zero, the masses can orbit in quite
stable ellipses about their common centre of gravity. This, of
course, is the acid test!

The program also has the useful capacity to demonstrate - or,
more precisely, underline - the conservation of momentum even
when it goes wrong - i.e. when a collision occurs.

Program Listing

   10 REM -R.V.Woodworth-Feb83-
   20 DIM ACC(20,1):DIM DAT(10,4):DIM ALSO(10,1):S=0:U=0:V=0
:H=0:I=0:J=0:D=0:B=0:C=0:N=0:P=0:Q=0:K=0:T=1
   30 E=0:F=0
   40 REM*************************************** INSTRUCTION
S-DESCRIPTION *******
   50 PRINT TAB(12,7)"NPARTICLE":PRINT TAB(15,8)"BY":PRINT T
AB(8,10)"R.V.Woodworth"
   60 PRINT"  This program plots the path of any "'"number o
f particles moving under their"'"mutual gravitational attrac
tions."
   70 PRINT"  It requires,for each particle,MASS,"'"START Co
ords,SPEED and ANGLE."
   80 PRINT"  The shorter the sampling interval,"'"the more 
accurate the approximations"'"used.(And the slower the plot.
)"'"Similarly the maths. breaks down if"'"they approach each
 other too closely."
   90 PRINT"  Newton's third law is however always"'"obeyed 
so MOMENTUM at least is "'"conserved."
  100 PRINT" Clearly the initial total momentum "'"must be z
ero if the particles ARE to"'"stay on the screen."
  110 PRINT"The units used are arbitary so data for"'"a sugg
ested first example are included"'"here."''
  120 PRINT" No.-2,M1-50,X1-600,Y1-600,V1-10,<0"''" M2-5,X2-
600,Y2-100,V2-100,<180"''
  130 REM******************************************* ACCEPT 
DATA ****************
  140 PRINT" To change T press T otherwise press"'"any key"
  150 IF GET$="T" INPUT"T IS PRESENTLY 1.PLEASE ENTER NEW VA
LUE.  T=",T
  160 INPUT"PLEASE ENTER THE NUMBER OF PARTICLES.    N="R:T=
T/10
  170 INPUT"PLEASE ENTER DATA FOR EACH MASS IN TURN":REPEAT
  180   PRINT"MASS NUMBER";(S+1):INPUT"MASS="M,"XPOS="X,"YPO
S="Y,"SPEED="V,"ANGLE(degs)="A:A=A*PI/180
  190 DAT(S,0)=M*100000:DAT(S,1)=X:DAT(S,2)=Y:DAT(S,3)=V*COS
A:DAT(S,4)=V*SINA:S=S+1:UNTIL S=R
  200 MODE4 
  210 REM******************************************* OUTER L
OOP(start again)*****
  220 S=0
  230 REM********************* (one mass-Ms-at a time)******
*

  240 REPEAT:M=DAT(S,0):X=DAT(S,1):Y=DAT(S,2):U=DAT(S,3):V=D
AT(S,4):H=0:PLOT69,X,Y:B=0:C=0
  250   REM*********************************************** I
NNER LOOP(to 310)********adds forces due to each other mass 
on Ms
  260   REPEAT:IF H=S GOTO 300
  270     N=DAT(H,0):I=DAT(H,1):J=DAT(H,2):D=(SQR((I-X)*(I-X
)+(J-Y)*(J-Y)))^3
  280     D=(SQR((I-X)*(I-X)+(J-Y)*(J-Y)))^3
  290     B=B+N*(I-X)/D:C=C+N*(J-Y)/D
  300   H=H+1:UNTIL H=R
  310   REM******* INNER ***********************************
*********************
  320   E=ACC(S,0):F=ACC(S,1):ACC(S,0)=B:ACC(S,1)=C:B=B+K*(B
-E):C=C+K*(C-F)
  330   P=U+B*T:Q=V+C*T:DAT(S,3)=P:DAT(S,4)=Q:ALSO(S,0)=X+U*
T+.5*B*T*T:ALSO(S,1)=Y+V*T+.5*C*T*T
  340 S=S+1:UNTILS=R
  350 REM********* CENTRE **********************************
*********************
  360 S=0
  370 REPEAT:DAT(S,1)=ALSO(S,0):DAT(S,2)=ALSO(S,1)
  380 S=S+1:UNTILS=R
  390 K=.5
  400 GOTO220
  410 REM********* OUTER ***********************************
**********************
  420 END

