
7. Snooker

General Description

This program sets a snooker table on the screen. The object of
the game is to pot the ball into any pocket in as few shots as
possible. The computer asks the user to specify the angle (0-360)
at which the ball should be shot, and then the force (1-10) at
which it should be hit. You are given eight shots before the
computer begins to mock your efforts!

Detailed Description

Lines 10 Set up envelope for sound.
20-70 Display title.
75-76 First part of delete trick.
80-160 GOTO procedures.
170-500 Sets up variables, draws snooker table on screen and

puts ball on table at random position.
510-880 Get direction and force from keyboard and then

calculate speed of ball, hit the ball and check to see if it has gone
in, also check to see if ball has hit the cushion, and if so make
ball bounce off cushion. Erase the old ball and put new ball on
table. Add friction to speed.

890-940 Gives rating.
950-990 The ball is still on the table after eight shots, give

poor rating.
1000-1110 Instructions.
1120-1150 Second part of delete trick
1160-1200 Credits.

Educational Notes

Like the Archery program, this simulation is useful in that it
forces the user to anticipate the effect of two variables utilised in
tandem. In the classroom it can be used to good effect to enable
youngsters to recognise angles. In the home it's simply
entertaining.

Program Listing

>
   10 ENVELOPE 1,1,0,0,0,10,20,30,70,-3,-1,-4,126,126
   20 MODE 7
   30 TITLE$=CHR$(130)+"SNOOKER"
   40 PRINTTAB(10,4);:VDU141:PRINTTITLE$
   50 PRINTTAB(10,5);:VDU141:PRINTTITLE$
   60 PRINTTAB(6,10);:VDU136:PRINT"PRESS SPACE TO START"
   70 A$=INKEY$(900)
   75 ONERRORGOTO80
   76 GOTO1000
   80 MODE 5:VDU23;8202;0;0;0;
   90 PROCset_up
  100 PROCmove_ball
  110 GCOL0,3
  120 IF SHOTS=8THENMODE7:PROCuseless
  130 IFPOCKETED=0THEN100
  140 MODE7
  150 PROCrating
  160 END
  170 DEF PROCset_up
  180 POCKETED=0
  190 SHOTS=0
  200 DRAG=0.97
  210 BOUNCE=-0.9
  220 XL=50:XM=1230
  230 YL=200:YM=810
  240 PXL=XL:PYL=YL+50
  250 PXM=XM-70:PYM=YM+10
  260 MI=(PXM+PXL)/2
  270 REM **** DRAW THE TABLE ****
  280 VDU19,2,2,0,0,0
  290 CLS:CLG
  300 GCOL 0,2
  310 MOVE XL,YL
  320 MOVE XM,YL
  330 PLOT 85,XM,YM
  340 DRAW XL,YM
  350 PLOT 85,XL,YL

  360 REM **** DRAW THE POCKETS **** 
  370 VDU5
  380 GCOL0,3
  390 MOVE XL,YL+30:PRINT"O"
  400 MOVE XL,YM:PRINT"O"
  410 MOVE XM-50,YM:PRINT"O"
  420 MOVE XM-50,YL+30:PRINT"O"
  430 MOVE MI,YL+30:PRINT"O"
  440 MOVE MI,YM:PRINT"O"

  450 REM **** DRAW BALL ON TABLE **** 
  460 BALL$="."
  470 GCOL0,3
  480 X=RND(PXM-400)+200:Y=RND(PYM-400)+300
  490 MOVE X,Y:PRINT BALL$
  500 ENDPROC
  510 DEF PROCmove_ball
  520 COLOUR 132
  530 VDU4
  540 PRINTTAB(0,0);SPC(80);TAB(0,0);
  550 INPUT"DIRECTION    ",ANGLE
  560 IF ANGLE<0ORANGLE>360VDU7:GOTO540
  570 INPUT"FORCE(0-10) ",F
  580 IF F<0ORF>10VDU7:GOTO540
  590 VDU5
  600 F=F*10
  610 REM *** CALCULATE SPEED *** 
  620 VX=F*SIN(ANGLE*0.0174533)
  630 VY=F*COS(ANGLE*0.0174533)
  640 REM *** NOW HIT THE BALL *** 
  650 SHOTS=SHOTS+1
  660 FOR T=1TO1000
  670   X1=X+VX:Y1=Y+VY 
  680   XI=INT(X1):YI=INT(Y1)
  690   REM *** HAS IT GONE IN *** 
  700   IF (ABS(YI-PYL)<30 OR ABS(YI-PYM)<30) AND (ABS(XI-PX
L)<30 OR ABS(XI-PXM)<30 OR ABS(XI-MI)<30) THEN POCKETED=1:EN
DPROC       
  710   REM *** CHECK FOR CUSHION ***
  720   IF X1<PXL THEN VX=BOUNCE*VX:X1=PXL+PXL-X1:SOUND0,1,2
,1
  730   IF XI>PXM THEN VX=BOUNCE*VX:X1=PXM+PXM+1-X1:SOUND0,1
,2,1
  740   IF Y1<PYL THEN VY=BOUNCE*VY:Y1=PYL+PYL-Y1:SOUND0,1,2
,1
  750   IF YI>PYM THEN VY=BOUNCE*VY:Y1=PYM+PYM+1-Y1:SOUND0,1
,2,1
  760   REM *** ERASE OLD BALL ***
  770   GCOL0,2
  780   IFXI<>INT(X)ORYI<>INT(Y)THENMOVEX,Y:PRINTBALL$
  790   REM *** RE-DRAW THE BALL *** 
  800   GCOL0,3
  810   MOVEX1,Y1:PRINTBALL$
  820   FORK=1TO40:NEXTK
  830   X=X1:Y=Y1
  840   REM ****** FRICTION ****** 
  850   VX=DRAG*VX:VY=DRAG*VY
  860   IFABS(VX)<0.05ANDABS(VY)<0.05THENENDPROC 
  870 NEXTT
  880 ENDPROC
  890 DEF PROCrating
  900 CLS
  910 PRINT''''"WELL DONE, YOU POT THE BALL IN"
  920 IF SHOTS=1THENPRINT''STR$(SHOTS);"SHOT,  YOU MUST BE S
TEVE DAVIS !!!":GOTO 940
  930 PRINT''SHOTS;" SHOTS"
  940 ENDPROC
  950 DEFPROCuseless
  960 CLS
  970 PRINT ''''"8 SHOTS AND STILL IT'S ON THE TABLE !!!"
  980 PRINT ''"TRY TAKING SOME LESSONS."
  990 END
 1000 MODE7
 1010 PRINTTAB(10,3);"INSTRUCTIONS"
 1020 PRINT"You must try and pot the ball in as"
 1030 PRINT"few shots as possible."
 1040 PRINT'"I will ask you for the angle you"
 1050 PRINT"wish the ball to go in and you must "
 1060 PRINT"type in an angle between 0 and 360."

 1070 PRINT'"I will then ask you to type in the force";
 1080 PRINT"and this must be between 1 and 10."
 1090 PRINT'"DO YOU WISH TO STOP THIS PROGRAM (Y/N)";
 1100 A$=GET$:IFA$="Y"THENEND
 1110 PRINTA$
 1120 PRINT'"PRESS THE FUNCTION KEY 'fo'"
 1130 *KEY0DELETE1000,1200|MRUN|M
 1140 IFINKEY(-33)=0THEN1140
 1150 END
 1160 ********************************
 1170 *          SNOOKER             *
 1180 *        WRITTEN BY            *
 1190 *     Andrew Pusey. FEB 1983   *
 1195 ********************************
 1200   DO NOT RENUMBER THIS PROGRAM

