

35. Periodic Table

General Description

The program displays the periodic table of the elements, in various groupings, with full categorisation of the element types, which both distinguishes it from simpler and shorter programs merely listing the elements, and accounts for the length of the program generally, with the multiple data lists such a categorisation requires. The program first presents a menu to the user, to allow a choice of the main display groupings, and the two test routines included in the program. Subsidiary menus are presented in the procedures accessed from the main menu to allow further choice of groupings and periods. The presentation of the elements in coloured tabular form makes them clear and easy to understand and remember.

The program as it stands, with full remark statements and the necessary spacing around the REMs to render the listing readable, will not run on a Model A, but will fit if you remove all remarks and cut down the variable names, which again are there for clarity. Note however that the PRINT statements are run together in places to fit the screen, so type the statements exactly as they appear in the listing.

The BREAK and ESCAPE keys are trapped in this program, but the auto-repeat is left connected, and the keyboard buffer is not flushed. This has implications for the test sections of the program, but allows quick scanning elsewhere in the program.

Detailed Description

Lines 10-250 Main program section calling routines selected from menu display.

260-1740 These procedures initialise the program, display

titles and the instructions (when requested from the menu), and provide the data for the program. Great care must be exercised to ensure the data is typed in correctly. The 999 which appears in the data statements after line 1420 acts as a data terminator, but the trailing zeroes are needed to prevent the READ statements in lines 660-700 from failing.

1750-2100 These lines contain the menu procedure, and the text display for the routines to display elements by periodic groupings, with a subsidiary menu for the choice of period. Note the keyboard validations in lines 1910-1960 and 2060-2080.

2110-2800 These lines have the procedure for display of chosen periods of the elements.

2810-3680 Procedures to display particular groups of elements as chosen in lines 3130-3160.

3690-end This sections contains the test sections of the program. The conversion routine in lines 4100-4160 means the program will accept answers both in lower and upper case. The correct answers are displayed when you have finished the test. Both tests are 20 questions at random.

Educational Note

These types of programs are good exercises in information retrieval for upper school youngsters. There is nothing in the program which cannot be found in a book, but the presentation in groupings and periods and the very interaction with the computer has tutorial value. The level of knowledge assumed is good O-level or A-level, and below this standard youngsters will not find it very meaningful. It provides a good quick reference for teachers, but must of course be located before it is needed to fulfil this function.

Program Listing

```

10 REM =====
20 REM   PERIODIC TABLE
30 REM
40 REM   BY CW FEB 1983
50 REM   .....
60 REM
70 DIM NAME$(103),SYMBOL$(103),GROUP%(8,8),PERIOD%(7,9),T
RANS%(3,11),RARE%(15),URAN%(15),TEST%(20),ANS$(20)

```

```

80 ON ERROR GOTO 4690
90 MODE7
100 *KEY10 OLD|MRUN|M
110 PROC_HEADER
120 PROC_SETUP
130 PROC_INSTR
140 REPEAT
150   PROC_MENU
160   ON SELECT% GOTO 170,180,190,200,210,220,230
170   PROC_INSTR: GOTO 230
180   PROC_DISP1: GOTO 230
190   PROC_DISP2: GOTO 230
200   PROC_DISP3: GOTO 230
210   PROC_TEST1: GOTO 230
220   PROC_TEST2: GOTO 230
230 UNTIL SELECT%=7
240 CLS
250 END
260 REM =====
270 REM   PROC_HEADER
280 REM .....
290 REM
300 DEF PROC_HEADER
310 CLS
320 PRINT TAB(14,8);CHR$141;CHR$130;"PERIODIC";TAB(14,9);C
HR$141;CHR$130;"PERIODIC"
330 PRINT TAB(15,11);CHR$141;CHR$130;"TABLE";TAB(15,12);CH
R$141;CHR$130;"TABLE"
340 PRINT TAB(11,14);CHR$134;"By C.W. Feb 1983"
350 ENDPROC
360 REM =====
370 REM   PROC_INSTR (INSTRUCTIONS)
380 REM .....
390 REM
400 DEF PROC_INSTR
410 CLS
420 PROC_TITLE
430 PRINT" This program displays the PERIODICTABLE o
f elements by periods and groups.It also lists the elemen
t names andsymbol for each element. Simple testsare inc
luded to test your knowledge of";
440 PRINT"the elements, symbols and the periodictable. A
ll the answers are included inthe program."
450 PRINT"" Every element has an atomic numberstarti
ng with Hydrogen with an atomicnumber of 1. The horizon
tal rows arecalled periods and the vertical columnsare ca
lled groups.";
460 PRINT" There are 7 periods and 8 groups."
470 PRINT TAB(0,23)"PRESS ANY KEY TO CONTINUE"
480 Z$=GET$
490 ENDPROC
500 REM =====
510 REM   PROC_TITLE
520 REM .....
530 REM
540 DEF PROC_TITLE
550 PRINT TAB(11,0);CHR$141;CHR$130;"PERIODIC TABLE"
560 PRINT TAB(11,1);CHR$141;CHR$130;"PERIODIC TABLE"
570 ENDPROC
580 REM =====
590 REM   PROC_SETUP
600 REM .....
610 REM
620 DEF PROC_SETUP
630 FOR I%=1 TO 103
640   READ NAME$(I%),SYMBOL$(I%)
650 NEXT I%
660 FOR I%=1 TO 8
670   FOR J%=1 TO 3

```

```

680     READ GROUP%(I%,J%)
690     NEXTJ%
700 NEXT I%
710 FOR I%=1 TO 7
720     FOR J%=1 TO 9
730         READ PERIOD%(I%,J%)
740     NEXT J%
750 NEXT I%
760 FOR I%=1 TO 3
770     FOR J%=1 TO 11
780         READ TRANS%(I%,J%)
790     NEXT J%
800 NEXT I%
810 FOR I%=1 TO 15
820     READ RARE%(I%)
830 NEXT I%
840 FOR I%=1 TO 15
850     READ URAN%(I%)
860 NEXT I%
870 DATA HYDROGEN,H,HELIUM,He
880 DATA LITHIUM,Li,BERYLLIUM,Be
890 DATA BORON,B,CARBON,C
900 DATA NITROGEN,N,OXYGEN,O
910 DATA FLUORINE,F,NEON,Ne
920 DATA SODIUM,Na,MAGNESIUM,Mg
930 DATA ALUMINIUM,Al,SILICON,Si
940 DATA PHOSPHOROUS,P,SULPHUR,S
950 DATA CHLORINE,Cl,ARGON,A
960 DATA POTASSIUM,K,CALCIUM,Ca
970 DATA SCANDIUM,Sc,TITANIUM,Ti
980 DATA VANADIUM,V,CHROMIUM,Cr
990 DATA MANGANESE,Mn,IRON,Fe
1000 DATA COBALT,Co,NICKEL,Ni
1010 DATA COPPER,Cu,ZINC,Zn
1020 DATA GALLIUM,Ga,GERMANIUM,Ge
1030 DATA ARSENIC,As,SELENIUM,Se
1040 DATA BROMINE,Br,KRYPTON,Kr
1050 DATA RUBIDIUM,Rb,STRONTIUM,Sr
1060 DATA YTTRIUM,Y,ZIRCONIUM,Zr
1070 DATA NIOBIUM,Nb,MOLYBDENUM,Mo
1080 DATA TECHNETIUM,Tc,RUTHENIUM,Ru
1090 DATA RHODIUM,Rh,PALLADIUM,Pd
1100 DATA SILVER,Ag,CADMIUM,Cd
1110 DATA INDIUM,In,TIN,Sn
1120 DATA ANTIMONY,Sb,TELLURIUM,Te
1130 DATA IODINE,I,XENON,Xe
1140 DATA CAESIUM,Cs,BARIUM,Ba
1150 DATA LANTHANUM,La,CERIUM,Ce
1160 DATA PRASEODYMIUM,Pr
1170 DATA NEODYMIUM,Nd,PROMETHEUM,Pm
1180 DATA SAMARIUM,Sm,EUROPIUM,Eu
1190 DATA GADOLINIUM,Gd,TERBIUM,Tb
1200 DATA DYSPROSIUM,Dy,HOLMIUM,Ho
1210 DATA ERBIUM,Er,THULIUM,Tm
1220 DATA YTTERBIUM,Yt,LUTECIUM,Lu
1230 DATA HAFNIUM,Hf,TANTALUM,Ta
1240 DATA TUNGSTEN,W,RHENIUM,Re
1250 DATA OSMIUM,Os,IRIDIUM,Re
1260 DATA PLATINUM,Pt,GOLD,Au
1270 DATA MERCURY,Hg,THALLIUM,Tl
1280 DATA LEAD,Pb,BISMUTH,Bi
1290 DATA POLONIUM,Po,ASTATINE,As
1300 DATA RADON,Rn,FRANCIUM,Fr,RADIUM,Re
1310 DATA ACTINIUM,Ac,THORIUM,Th
1320 DATA PROACTINIUM,Pa,URANIUM,U
1330 DATA NEPTUNIUM,Np,PLUTONIUM,Pu
1340 DATA AMERICIUM,Am,CURIUM,Cm
1350 DATA BERKELIUM,Bk,CALIFORNIUM,Cm
1360 DATA EINSTEINIUM,Es,FERMIUM,Fm

```

```

1370 DATA MENDELEVIUM,Mv,NOBELIUM,No
1380 DATA LAWRENCIUM,Lw
1390 REM .....
1400 REM ELEMENTS IN GROUPS
1410 REM .....
1420 DATA 1,3,11,19,37,55,87,999
1430 DATA 4,12,20,38,56,88,999,0
1440 DATA 5,13,31,49,81,999,0,0
1450 DATA 6,14,32,50,82,999,0,0
1460 DATA 7,15,33,51,83,999,0,0
1470 DATA 8,16,34,52,84,999,0,0
1480 DATA 9,17,35,53,85,999,0,0
1490 DATA 10,18,36,54,86,999,0,0
1500 REM .....
1510 REM ELEMENTS IN PERIODS
1520 REM .....
1530 DATA 1,2,999,0,0,0,0,0
1540 DATA 3,4,5,6,7,8,9,10,999
1550 DATA 11,12,13,14,15,16,17,18,999
1560 DATA 19,20,31,32,33,34,35,36,999
1570 DATA 37,38,49,50,51,52,53,54,999
1580 DATA 55,56,81,82,83,84,85,86,999
1590 DATA 87,88,999,0,0,0,0,0
1600 REM .....
1610 REM TRANSITIONAL ELEMENTS
1620 REM .....
1630 DATA 21,22,23,24,25,26,27,28,29,30,999
1640 DATA 39,40,41,42,43,44,45,46,47,48,999
1650 DATA 72,73,74,75,76,77,78,79,80,999,0
1660 REM .....
1670 REM RARE EARTHS
1680 REM .....
1690 DATA 57,58,59,60,61,62,63,64,65,66,67,68,69,70,71
1700 REM .....
1710 REM TRANS-URANICS
1720 REM .....
1730 DATA 89,90,91,92,93,94,95,96,97,98,99,100,101,102,103
1740 ENDPROC
1750 REM =====
1760 REM PROC_MENU
1770 REM .....
1780 REM
1790 DEF PROC_MENU
1800 CLS
1810 PROC_TITLE
1820 Z%=0
1830 PRINT'"No. Title"
1840 PRINT"1. Instructions"
1850 PRINT"2. Show Elements by Period"
1860 PRINT"3. Show Elements by Group"
1870 PRINT"4. Show Element Names and Symbols"
1880 PRINT"5. Test1 - Symbols and Elements"
1890 PRINT"6. Test2 - Names for Symbols"
1900 PRINT"7. END"
1910 PRINT'"Choose your selection from the menu. Andtype
a number from 1 to 7"
1920 PRINT'"CHR$134;"TYPE THE NUMBER OF YOUR CHOICE";CHR$13
2;:INPUT SELECT$
1930 IF SELECT$<"1" OR SELECT$>"7" THEN GOTO 1800
1940 SELECT%=VAL(SELECT$)
1950 IF SELECT%<1 OR SELECT%>7 THEN GOTO 1800
1960 ENDPROC
1970 REM =====
1980 REM PROC_DISP1(ELEMENTS BY PERIOD)
1990 REM .....
2000 REM
2010 DEF PROC_DISP1
2020 CLS:PROC_TITLE
2030 PRINT'" There are seven periods (1 to 7). Thehigher

```

```

the period, the higher the AtomicNumber of its elements.
The chemicalproperties of elements vary gradually";
2040 PRINT"across the period. The elements withlower nu
mbers in a period are metals, asthe atomic number increases
the elementsbecome non-metallic and end with an inertgas."
2050 PRINT'"Choose the period you want to look at.Type i
n the number (between 1 and 7)."'
2060 INPUT Z$
2070 IF Z$<"1" OR Z$>"7" THEN GOTO 2020
2080 Z%=VAL(Z$)
2090 PROC_PERIOD(Z%)
2100 ENDPROC
2110 REM =====
2120 REM PROC_PERIOD (DISPLAY A PERIOD)
2130 REM .....
2140 REM
2150 DEF PROC_PERIOD(A%)
2160 CLS
2170 PROC_TITLE
2180 PRINT'"PERIOD      ";A%
2190 PRINT'"GROUP";
2200 IF A%=1 THEN PRINT TAB(9);"1";TAB(14)"8":GOTO 2220
2210 FOR I%=1 TO 7:PRINT TAB(9+I%*4);I%+1;:NEXT I%
2220 PRINT'CHR$129;"At. No";
2230 I%=1:REPEAT
2240   PRINT TAB(9+(I%-1)*4);PERIOD%(A%,I%);
2250   I%=I%+1
2260 UNTIL I%=9 OR PERIOD%(A%,I%)=999
2270 PRINT'CHR$131;"ELEMENT";
2280 I%=1:REPEAT
2290   PRINTTAB(9+(I%-1)*4);SYMBOL$(PERIOD%(A%,I%));
2300   I%=I%+1
2310 UNTIL I%=9 OR PERIOD%(A%,I%)=999
2320 IF A%<4 THEN GOTO 2720
2330 IF A%=7 GOTO 2460
2340 PRINT'"TRANSITIONAL ELEMENTS"
2350 PRINT CHR$129;
2360 I%=1:REPEAT
2370   PRINTTAB(9+(I%-1)*4);TRANS%(A%-3,I%);
2380   I%=I%+1
2390 UNTIL I%=9 OR PERIOD%(A%,I%)=999
2400 PRINT'CHR$131
2410 I%=1:REPEAT
2420   PRINT TAB(9+(I%-1)*4);SYMBOL$(TRANS%(A%-3,I%));
2430   I%=I%+1
2440 UNTIL I%=9 OR PERIOD%(A%,I%)=999
2450 IF A%<6 THEN GOTO 2720
2460 IF A%=6 THEN PRINT'"RARE EARTHS" ELSE PRINT'"TRANS-U
RANICS"
2470 PRINT CHR$129
2480 I%=1:REPEAT
2490   IF A%=6 THEN PRINT TAB(9+(I%-1)*4);RARE%(I%);
2500   IF A%=7 THEN PRINT TAB(9+(I%-1)*4);URAN%(I%);
2510   I%=I%+1
2520 UNTIL I%=9
2530 PRINT'CHR$131;
2540 I%=1:REPEAT
2550   IF A%=6 THEN PRINT TAB(9+(I%-1)*4);SYMBOL$(RARE%(I%
);
2560   IF A%=7 THEN PRINT TAB(9+(I%-1)*4);SYMBOL$(URAN%(I%
);
2570   I%=I%+1
2580 UNTIL I%=9
2600 PRINT'CHR$129
2610 I%=9:REPEAT
2620   IF A%=6 THEN PRINT TAB(9+(I%-9)*4);RARE%(I%);
2630   IF A%=7 THEN PRINT TAB(9+(I%-9)*4);URAN%(I%);
2640   I%=I%+1
2650 UNTIL I%=16

```

```

2660 PRINT'CHR$131
2670 I%=9:REPEAT
2680 IF A%=6 THEN PRINT TAB(9+(I%-9)*4);SYMBOL$(RARE%(I%
);
2690 IF A%=7 THEN PRINT TAB(9+(I%-9)*4);SYMBOL$(URAN%(I%
);
2700 I%=I%+1
2710 UNTIL I%=16
2720 PRINTTAB(0,22)"Do you want the names of the elements"
2730 PRINT"TYPE Y OR N ";:Z$=GET$
2740 IF Z$<>"Y" AND Z$<>"N" THEN PRINT TAB(0,23);SPC(35): G
OTO 2720
2750 IF Z$="N" THEN GOTO 2800
2760 TITLE$="PERIOD"+" "+STR$(A%)
2770 IF A%<7 THEN J%=PERIOD%(A%+1,1)-1
2780 IF A%=7 THEN J%=103
2790 PROC_ELEMENT(PERIOD%(A%,1),J%,TITLE$)
2800 ENDPROC
2810 REM =====
2820 REM PROC_ELEMENT LIST ELEMENTS
2830 REM .....
2840 REM
2850 DEF PROC_ELEMENT(A%,B%,Z$)
2860 LOCAL I%,J%
2870 J%=A%
2880 CLS:PROC_TITLE:I%=1
2890 PRINT Z$
2900 PRINT "Atomic no. ";TAB(12);"Name";TAB(25);"Symbol"
2910 REPEAT
2920 PRINT TAB(3);CHR$129;J%;TAB(11);CHR$134;NAME$(J%);TA
B(27);CHR$131;SYMBOL$(J%)
2930 J%=J%+1:I%=I%+1
2940 UNTIL I%=20 OR J%=B%+1
2950 IF J%=B%+1 THEN GOTO 2990
2960 PRINT TAB(0,24)"PRESS ANY KEY TO CONTINUE";
2970 A$=GET$
2980 GOTO 2880
2990 PRINT TAB(0,24)"PRESS ANY KEY TO FINISH";
3000 A$=GET$
3010 ENDPROC
3020 REM =====
3030 REM PROC_DISP2 ELEMENTS BY GROUP
3040 REM .....
3050 REM
3060 DEF PROC_DISP2
3070 CLS:PROC_TITLE
3080 PRINT"" There are 8 groups of elements (1 - 8)there
are also the transition elements,rare earths and trans-uran
ic elements."
3090 PRINT"" These are all displayed here. To listthe ele
ments in these groups type 1 to 8for the groups, T for th
e transitionelements, R for the rare earths andU for t
he trans-uranic elements."
3100 PRINT""TYPE YOUR CHOICE (1-8,T,R,U)"
3110 Z$=GET$
3120 IF (Z$<"1" OR Z$>"8") AND Z$<>"R" AND Z$<>"T" AND Z$<>
"U" THEN GOTO 3070
3130 IF Z$="R" THEN PROC_RARE: GOTO 3170
3140 IF Z$="T" THEN PROC_TRANS: GOTO 3170
3150 IF Z$="U" THEN PROC_URAN: GOTO 3170
3160 Z%=VAL(Z$): PROC_GROUP(Z%)
3170 ENDPROC
3180 REM =====
3190 REM PROC_RARE DISPLAY RARE EARTHS
3200 REM .....
3210 REM
3220 DEF PROC_RARE
3230 TITLE$="RARE EARTHS"
3240 PROC_ELEMENT(57,71,TITLE$)

```

```

3250 ENDPROC
3260 REM
3270 REM PROC_TRANS DISPLAY TRANSITION ELEMENTS
3280 REM
3290 DEF PROC_TRANS
3300 TITLE$="TRANSITION ELEMENTS - PERIOD 4"
3310 PROC_ELEMENT(21,30,TITLE$)
3320 TITLE$="TRANSITION ELEMENTS - PERIOD 5"
3330 PROC_ELEMENT(39,48,TITLE$)
3340 TITLE$="TRANSITION ELEMENTS - PERIOD 6"
3350 PROC_ELEMENT(72,80,TITLE$)
3360 ENDPROC
3370 REM =====
3380 REM PROC_URAN DISPLAY TRANS-URANIC ELEMENTS
3390 REM .....
3400 REM
3410 DEF PROC_URAN
3420 TITLE$="TRANS-URANIC ELEMENTS"
3430 PROC_ELEMENT(89,103,TITLE$)
3440 ENDPROC
3450 REM =====
3460 REM PROC_GROUP DISPLAY ELEMENT IN A GROUP
3470 REM .....
3480 REM
3490 DEF PROC_GROUP(X%)
3500 CLS:PROC_TITLE
3510 PRINT'"GROUP      ";X%
3520 PRINT"Atomic No.";TAB(12);"Name";TAB(25);"Symbol"
3530 I%=1
3540 REPEAT
3550   PRINT TAB(3);CHR$129;GROUP%(X%,I%);TAB(11);CHR$131;N
AME$(GROUP%(X%,I%));TAB(26);CHR$134;SYMBOL$(GROUP%(X%,I%))
3560   I%=I%+1
3570 UNTIL GROUP%(X%,I%)=999
3580 PRINT TAB(0,24)"PRESS ANY KEY TO FINISH";
3590 A$=GET$
3600 ENDPROC
3610 REM =====
3620 REM PROC_DISP3 DISPLAY ELEMENTS
3630 REM .....
3640 REM
3650 DEF PROC_DISP3
3660 TITLE$="ALL ELEMENTS BY ATOMIC NUMBER"
3670 PROC_ELEMENT(1,103,TITLE$)
3680 ENDPROC
3690 REM =====
3700 REM PROC_TEST1 TEST SYMBOLS FOR ELEMENTS
3710 REM .....
3720 REM
3730 DEF PROC_TEST1
3740 CLS: PROC_TITLE
3750 PRINT'"TEST1 SYMBOLS FOR ELEMENTS"
3760 PRINT'" In this test the names of different element
s are shown on the screen. You have to type in the cor
rect chemical symbol for each element."
3770 PRINT'" For example if the element displayed is";CHR
$130;"NITROGEN";CHR$135;"your reply should be";CHR$131;"N";C
HR$135;". "
3780 PRINT'" If the chemical symbol for an element is 2 ch
aracters, for example";CHR$130;"Magnesium";CHR$135;". the sym
bol should be typed as ";CHR$131;"Mg";CHR$135;". But the pro
gram will accept";CHR$133;"MG";CHR$135;"or";CHR$134;"mg."
3790 PRINT'"There are 20 questions in this test."
3800 PRINT TAB(0,24)"PRESS ANY KEY TO CONTINUE";
3810 A$=GET$
3820 FOR I%=1 TO 20
3830   TEST%(I%)=RND(103)
3840   FOR J%=0 TO I%-1
3850     IF TEST%(J%)=TEST%(I%) THEN TEST%(I%)=RND(103): J%

```



```

=0
3860 NEXT J%
3870 NEXT I%
3880 SCORE%=0
3890 CLS:PROC_TITLE
3900 FOR I%=1 TO 20
3910 PRINT NAME$(TEST%(I%));TAB(20);"Symbol ";
3920 INPUT ANS$(I%)
3930 NEXT I%
3940 PRINT "PRESS ANY KEY FOR ANSWERS";:A$=GET$
3950 CLS:PROC_TITLE
3960 FOR I%=1 TO 20
3970 PRINT NAME$(TEST%(I%));TAB(15);CHR$130;ANS$(I%);
3980 Ans$=FNANS(ANS$(I%))
3990 PRINT TAB(24);CHR$131;SYMBOL$(TEST%(I%));
4000 IF Ans$=SYMBOL$(TEST%(I%)) THEN PRINT TAB(30);CHR$13
4;"CORRECT";:SCORE%=SCORE%+1 ELSE PRINT TAB(30);CHR$129;"WRO
NG";
4010 PRINT
4020 NEXT I%
4030 PRINT "SCORE = ";SCORE%;" OUT OF 20"
4040 PRINT "PRESS ANY KEY TO CONTINUE";:A$=GET$
4050 ENDPROC
4060 REM =====
4070 REM CONVERT ANSWER - FNANS
4080 REM .....
4090 REM
4100 DEF FNANS(A$)
4110 LOCAL B$,Y%
4120 IF ASC(LEFT$(A$,1))>90 THEN B$=CHR$(ASC(LEFT$(A$,1))-3
2) ELSE B$=LEFT$(A$,1)
4130 IF LEN(A$)=1 THEN GOTO 4160
4140 FOR Y%=2 TO LEN(A$):IF ASC(MID$(A$,Y%,1))<97 THEN B$=B
$+CHR$(ASC(MID$(A$,Y%,1))+32) ELSE B$=B$+MID$(A$,Y%,1)
4150 NEXT Y%
4160 =B$
4170 REM =====
4180 REM PROC_TEST2 NAMES FOR SYMBOLS
4190 REM .....
4200 REM
4210 DEF PROC_TEST2
4220 CLS:PROC_TITLE
4230 PRINT "TEST 2 ELEMENT NAMES FOR SYMBOLS"
4240 PRINT " In this test the symbols of differentelement
s are shown on the screen. You have to type in the cor
rect elementname."
4250 PRINT " For example if the symbol displayed is";CHR
$130;"Li";CHR$135;"", your reply should be";CHR$131;"Lithium"
;CHR$135;"."
4260 PRINT " The program will accept answers in both ca
pital and ordinary letters."
4270 PRINT " There are 20 questions in this test."
4280 PRINT TAB(0,24)"PRESS ANY KEY TO CONTINUE";
4290 A$=GET$
4300 FOR I%=1 TO 20
4310 TEST%(I%)=RND(103)
4320 FOR J%=0 TO I%-1
4330 IF TEST%(J%)=TEST%(I%) THEN TEST%(I%)=RND(103):J%=
0
4340 NEXT J%
4350 NEXT I%
4360 SCORE%=0
4370 CLS:PROC_TITLE
4380 FOR I%=1 TO 20
4390 PRINT SYMBOL$(TEST%(I%));TAB(19);"Name ";
4400 INPUT ANS$(I%)
4410 NEXT I%
4420 PRINT "PRESS ANY KEY FOR ANSWERS";:A$=GET$
4430 CLS:PROC_TITLE

```

```

4440 FOR I%=1 TO 20
4450     PRINT SYMBOL$(TEST%(I%));TAB(5);CHR$130;ANS$(I%);
4460     Ans$=FNCONV(ANS$(I%))
4470     PRINT TAB(19);CHR$131;NAME$(TEST%(I%));
4480     IF Ans$=NAME$(TEST%(I%)) THEN PRINT TAB(35);CHR$134;
"YES";:SCORE%=SCORE%+1 ELSE PRINT TAB(35);CHR$129;"NO";
4490     PRINT
4500 NEXT I%
4510 PRINT"SCORE = ";SCORE%;" OUT OF 20"
4520 PRINT "PRESS ANY KEY TO CONTINUE";:A$=GET$
4530 ENDPROC
4540 REM =====
4550 REM CONVERSION FUNCTION
4560 REM .....
4570 REM
4580 DEF FNCONV(A$)
4590 LOCAL B$,Y%
4600 B$=""
4610 FOR Y%=1 TO LEN(A$)
4620     IF ASC(MID$(A$,Y%,1))>90 THEN B$=B$+CHR$(ASC(MID$(A$
,Y%,1))-32) ELSE B$=B$+MID$(A$,Y%,1)
4630 NEXT Y%
4640 =B$
4650 REM =====
4660 REM   whos a naughty boy
4670 REM .....
4680 REM
4690 IF ERR=17 GOTO 110
4700 MODE7
4710 REPORT:PRINT" at line ";ERL
4720 END

```

Software

If you are interested in cassette versions of the programs featured in this book, please write for details to:

Dept. S
Century Publishing
76, Old Compton Street
London, W1V 5PA

Available in Century's Science & Technology series

DICTIONARY OF NEW INFORMATION TECHNOLOGY

A.J. Meadows, M. Gordon and A. Singleton

WHAT TO BUY FOR BUSINESS: A HANDBOOK OF NEW
OFFICE TECHNOLOGY

John Derrick and Phillip Oppenheim

THE ELECTRONIC MAIL HANDBOOK

Stephen Connell and Ian A. Galbraith

THE WAY THE NEW TECHNOLOGY WORKS

Ken Marsh

MIND

David A. Taylor

In association with 'Personal Computer World'

MICROCOMPUTING FOR BUSINESS: A USER'S GUIDE

edited by Dick Olney

THE MICROCOMPUTER HANDBOOK: A BUYER'S GUIDE

edited by Dick Olney

THE SPECTRUM HANDBOOK

Tim Langdell

35 PROGRAMS FOR THE DRAGON 32

Tim Langdell

THE INTIMATE MACHINE

Neil Frude

CENTURY COMPUTER PROGRAMMING COURSE

Peter Morse and Ian Adamson

