
11. Cosmic Nim

General Description

When this program is run, rows of unpleasant alien heads appear
on the left of the screen. The idea of the game is to play against
the computer, destroying as many of the heads as you like, but
endeavouring to manoeuvre the computer into the position where
it is left with run remaining head. If you have problems running
this on a Model A, remove all spaces.

Detailed Description

Lines 1-2 When the program is first run, there will be no
error and the commands in Line 2 will be executed. However,
when the instructions have been read, the instructions will then
be deleted (Lines 1050-1200) and this will cause an error in Line
2 the next time the program is run, so it will jump to Line 10.

Do not press the Escape key during the program's execution
because this will cause it to jump to Line 10 and it might run out
of memory at 10, so it will get into a continuous loop. To get out
of this press Break key and then type OLD.

10-210 Sets variables, colours and characters
220-260 Operates the procedures.
270-440 Displays heads and whether you have won or lost.
450-480 Is there only 1 head left? If so, user has won.
490-590 Is there only 1 row left? If so, leave row with 1 and

say the computer has won.
600-620 Procedure to take away heads.
630 If all rows are there take 1 away from Random row.
640-660 Two rows are left, redefine rows for computer as D

(X) where X is from 1-2, but leave old rows there.
670-690 If one row has 1 then take everything away from

other row and say that the computer has won.
700 2 left in each row, take 1 away from Random row.
710-720 Make both rows have same amount of heads.
830-1040 Move alien up right-hand side of screen and fire

torpedoes.
1050-1140 Instructions.
1150-1180 Redefine function key fo to delete the instructions

and then run program again e.g. Delete 1050, 1200 (carriage
return), Run (carriage return). See description of lines 1-2 for
more details.
1181-1189 Credits.

Educational Notes

Whilst enormously entertaining in its own right, Cosmic Nim is
a taxing logical exercise which, yet again, seems to defeat vast
hordes of adults and present no problems for the majority of
students.

Program Listing

    1 ONERRORGOTO10
    2 MODE7:GOTO1050
   10 FL=0:R=0:TA=0:FA=0:MODE5
   40 VDU19,0,2,0,0,0
   50 VDU23,240,31,63,127,127,115,115,127,127
   60 VDU23,241,127,63,63,28,14,15,7,0
   70 VDU23,242,248,252,254,254,206,206,254,254
   80 VDU23,243,254,252,252,56,112,240,224,0
   90 VDU23,244,31,63,127,127,255,255,255,255
  100 VDU23,245,255,255,127,127,127,63,31,7
  110 VDU23,246,192,224,240,240,152,156,252,240

  120 VDU23,247,240,240,240,0,240,240,224,128
  130 VDU19,2,7,0,0,0:VDU19,3,6,0,0,0
  140 VDU23,248,3,7,15,31,60,252,255,31
  150 VDU23,249,31,255,252,60,252,255,31
  160 VDU23,250,255,254,252,248,248,248,254,254
  170 VDU23,251,254,254,248,248,248,252,254,255
  180 G=0
  190 DIMN(3),D(2)
  200 CW$="":H=0
  210 N(1)=3:N(2)=4:N(3)=5
  220 PROC_DISPLAY
  230 PROC_HUMAN_MOV
  240 PROC_DISPLAY
  250 PROC_COMPUTER_MOV
  260 END
  270 DEFPROC_DISPLAY
  280 IFFL=1 PROC_ALIEN
  290 FL=1
  300 CLS
  310 PRINTTAB(6,2);"COSMIC NIM"
  320 FORT=1TO3
  330   COLOURT
  340   PRINTTAB(0,T*3+5);T
  350   IFN(T)=0THEN420
  360   FORY=1TON(T)
  370     PRINTTAB(Y*2,T*3+5);CHR$(240+FA);CHR$(242+FA)
  380     PRINTTAB(Y*2,T*3+6);CHR$(241+FA);CHR$(243+FA)
  390     IFRND(2)=1THENFA=4ELSEFA=0
  400   NEXTY
  410   PRINT'''CW$:IFCW$<>""END
  420 NEXTT
  430 COLOUR2
  440 ENDPROC
  450 DEFPROC_COMPUTER_MOV
  460 H=0:J=0
  470 FORT=1TO3:H=H+N(T):NEXTT
  480 IFH=1THENPRINT"YOU WIN":END
  490 FORT=1TO3
  500   IFN(T)<>0THENJ=J+1
  510 NEXTT
  520 IFJ=1THEN450
  530 GOTO630
  540 FORT=1TO3
  550   IFN(T)<>0THEN570
  560 NEXT
  570 TA=N(T)-1:R=T:CW$="I WIN"
  580 PROC_TAKE_AWAY
  590 GOTO220
  600 DEFPROC_TAKE_AWAY
  610 N(R)=N(R)-TA
  620 ENDPROC
  630 IFJ=3R=RND(3):TA=1:GOTO580
  640 FORT=1TO3
  650   IFN(T)<>0G=G+1:D(G)=T
  660 NEXTT
  670 IFN(D(2))=1R=D(1):TA=N(D(1)):CW$="I WIN":GOTO580
  680 G=0
  690 IFN(D(1))=1R=D(2):TA=N(D(2)):CW$="I WIN":GOTO580
  700 IFRND(2)=2ANDN(D(1))=N(D(2))R=D(2):TA=1:GOTO580ELSEIFN
(D(1))=N(D(2))R=D(1):TA=1:GOTO580
  710 IFN(D(1))<N(D(2))R=D(2):TA=N(D(2))-N(D(1)):GOTO580
  720 IFN(D(1))>N(D(2))R=D(1):TA=N(D(1))-N(D(2)):GOTO580
  730 DEFPROC_HUMAN_MOV
  740 PRINTTAB(3,20);"ROW";
  750 A$=GET$:R=VAL(A$):IFR<1ORR>3VDU7:GOTO740
  760 PRINTA$
  770 PRINTTAB(3,21);"NO.";
  780 A$=GET$:TA=VAL(A$)
  790 PRINTA$

  800 IFN(R)-TA<0ORTA=0VDU7:GOTO740
  810 PROC_TAKE_AWAY
  820 ENDPROC
  830 DEFPROC_ALIEN
  840 FORY=20TOR*3+5STEP-1
  850   PRINTTAB(17,Y);
  860   VDU248,250,8,8,10,249,251
  870   PRINTTAB(17,Y);
  880   SOUND1,-15,123,1
  890   A$=INKEY$(10)
  900   VDU32,32,8,8,10,32,32
  910 NEXTY
  920 Y=R*3+5
  930 FORX=15TO2+2*N(R)STEP-1
  940   PRINTTAB(X+1,Y);" "
  950   PRINTTAB(X+1,Y+1);" "
  960   PRINTTAB(X,Y);"{" 
  970   PRINTTAB(X,Y+1);"{" 
  980   SOUND0,-15,5,1
  990   A$=INKEY$(4)
 1000 NEXT
 1010 PRINTTAB(X+1,Y);" "
 1020 PRINTTAB(X+1,Y+1);" "
 1030 A$=INKEY$(100)
 1040 ENDPROC
 1050 PRINTTAB(15);"INSTRUCTIONS"
 1060 PRINT'''"THE IDEA OF THIS GAME IS TO DESTROY"
 1070 PRINT"THE ALIEN HEADS ON THE LEFT OF THE"
 1080 PRINT"SCREEN. BUT IF THE COMPUTER MAKES YOU"
 1090 PRINT"DESTROY THE LAST HEAD,THEN IT WINS."
 1100 PRINT"SO IF YOU WANT TO WIN, YOU MUST LEAVE"
 1110 PRINT"IT WITH THE LAST HEAD."
 1120 PRINT'"DO YOU WISH TO STOP THIS PROGRAM (Y/N)";
 1130 A$=GET$
 1140 IFA$="Y"THENEND
 1150 PRINT'"PRESS FUNCTION KEY 'f0' TO START."
 1160 *KEY0DELETE1050,1200|MRUN|M
 1170 IFINKEY(-33)=0THEN1170
 1180 END
 1181  *******************************
 1182  *      COSMIC NIM             *
 1183  *      WRITTEN BY             *
 1184  *    Andrew Pusey. FEB 1983   *
 1185  *******************************
 1186  
 1187     ON NO ACCOUNT RENUMBER THIS
 1188          PROGRAM.
 1189  

